

## Calcul sélectif de termes quelconques dans la solution générale d'un problème planétaire à l'ordre 1 des masses

Luc Duriez

Laboratoire d'Astronomie de l'Université de Lille 1 et IMCCE de l'Observatoire de Paris

[Luc.Duriez@univ-lille1.fr](mailto:Luc.Duriez@univ-lille1.fr)  
<http://www.imcce.fr>

Dernière révision Avril 2007

Il s'agit ici d'une application directe de la méthode de perturbations planétaires définie comme "théorie générale planétaire" dans le [Cours de Mécanique Céleste](#). C'est aussi une application directe et pratique des techniques de développement des perturbations des planètes détaillées dans le document intitulé [Développements de Mécanique Céleste avec Maple](#)

Dans ce type de solution, on suppose bien connues les valeurs des moyens mouvements moyens des planètes, notés  $N$ . On les relie aux moyens mouvements osculateurs  $n$  par l'introduction d'une variable  $p$  telle que  $n = N(1 + p)$ . On relie de même les demi-grands axes osculateurs  $a$  à des quantités analogues  $A$  telles que l'on ait pour chaque planète :

$$n^2 a^3 = N^2 A^3 = k(M + m). \text{ On a alors aussi : } a = A(1 + p)^{(-2/3)}$$

La variable  $p$  peut alors remplacer  $a$  ou  $n$ , avec par exemple la nouvelle équation de Lagrange :

$$\frac{dp}{(1 + p) dt} = -\frac{3 da}{2 a dt}$$

Si  $N$  est le moyen mouvement moyen, la longitude moyenne  $L$  doit s'écrire :  $L = N t + q$  où la variable  $q$  ne doit pas comporter de terme proportionnel au temps et vérifie l'équation  $\frac{dL}{dt} = N + \frac{dq}{dt}$ . Comme on a aussi l'équation de Lagrange entre  $L$  et la longitude

moyenne à l'origine  $\lambda_0$ :  $\frac{dL}{dt} = n + \frac{d\lambda_0}{dt}$ , on a finalement :

$$\frac{dq}{dt} = Np + \frac{d\lambda_0}{dt}$$

et on doit introduire dans  $p$  une constante d'intégration particulière  $p_0$  telle que  $Np_0$  annule tout terme constant qui pourrait se trouver dans l'expression de  $\frac{d\lambda_0}{dt}$ .

Les autres variables utilisées pour représenter les mouvements planétaires sont  $x = e \exp(I\varpi)$  et  $y = \sin(i/2) \exp(I\Omega)$ , dont on a déjà explicité les équations de Lagrange dans la procédure [EqLag](#). (Dans cette procédure le calcul de  $dL/dt$  correspond en fait à  $dq/dt$  ici : il y manque  $N$ )

Le principe de la méthode d'intégration des équations, mises sous la forme matricielle :

$$\frac{dV}{dt} = \varepsilon \Lambda(V, t)$$

où  $\varepsilon$  est une petite quantité de l'ordre des masses perturbatrices, est de chercher une solution de la forme

$$V = V_0(t) + \Delta V(V_0, t)$$

où  $V_0(t)$  est la solution générale du système d'équations réduit à sa partie séculaire (partie autonome ou indépendante de  $t$ ) et où  $\Delta V(V_0, t)$  est la solution particulière du système d'équations contenant tous les termes périodiques:

$$\frac{dV_0}{dt} = [\varepsilon \Lambda(V_0)] \quad \text{et} \quad \frac{\partial}{\partial t} \Delta V = \{\varepsilon \Lambda(V_0(t), t)\}$$

Ce système est limité ici à l'ordre 1 des masses, mais pourrait être exprimé et résolu à des ordres supérieurs. On ne donne ci-dessous que la façon d'obtenir la solution d'ordre 1 : On cherche une solution à très longue périodes du système séculaire puis on donne un exemple de report de cette solution dans un terme périodique

## 1. Initialisations

### Initialisations générale

```
> restart:
> chemin:='Maple/';
> read cat(chemin,'Serutil.pr'):
> readlib(poisson):readlib(mtaylor):
chemin := Maple/
```

### Initialisation pour les problèmes de type planétaire;

```
> read cat(chemin,'cosS.m'):
> read cat(chemin,'kepler20.m'):
> read cat(chemin,'COSSYmla10.m'):
> read cat(chemin,'DELTACOSSYmla5.m'):
> read cat(chemin,'Xnm10.m');
> read cat(chemin,'COSSXij10.m');
> Dmax:=6;
> ASR:=TRONC(ASR,Dmax,[X,Xb]): PRINTRONC(ASR,3,[X,Xb]);
> RSA:=TRONC(RSA,Dmax,[X,Xb]): PRINTRONC(RSA,3,[X,Xb]);
> THETA:=TRONC(THETA,Dmax,[X,Xb]): PRINTRONC(THETA,3,[X,Xb]);
> THETAB:=TRONC(THETAB,Dmax,[X,Xb]): PRINTRONC(THETAB,3,[X,Xb]);
> ASRj:=subs(X=X[j],Xb=Xb[j],ASR): ASRi:=subs(j=i,ASRj):
> RSAi:=subs(X=X[i],Xb=Xb[i],RSA): RSAj:=subs(i=j,RSAi):
> THETAi:=subs(X=X[i],Xb=Xb[i],THETA):
> THETAj:=subs(i=j,THETAi):
> THETABi:=subs(X=X[i],Xb=Xb[i],THETAB):
```

Dev<sub>2</sub>

- > THETABj:=subs(i=j,THETABi):
- > varXYij:=[X[i],Xb[i],X[j],Xb[j],Y[i],Yb[i],Y[j],Yb[j]]:
- > varXij:=[X[i],Xb[i],X[j],Xb[j]]:
- > varxyij:=[x[i],xb[i],x[j],xb[j],y[i],yb[i],y[j],yb[j]]:

$$\begin{aligned}
 & D_{max} := 6 \\
 & \frac{81}{80} X^6, + \dots +, \frac{9}{16} X^3 - \frac{1}{16} X^2 Xb - \frac{1}{16} X Xb^2 + \frac{9}{16} Xb^3 + \frac{1}{2} X^2 + \frac{1}{2} Xb^2 + \frac{1}{2} X + \frac{1}{2} Xb + 1 \\
 & - \frac{27}{160} X^6, + \dots +, - \frac{3}{16} X^3 + \frac{3}{16} X^2 Xb + \frac{3}{16} X Xb^2 - \frac{3}{16} Xb^3 - \frac{1}{4} X^2 + \frac{1}{2} X Xb - \frac{1}{4} Xb^2 - \frac{1}{2} X - \frac{1}{2} Xb + 1 \\
 & \frac{117649}{46080} X^6, + \dots +, \frac{4}{3} X^3 - \frac{5}{4} X^2 Xb - \frac{1}{12} Xb^3 + \frac{9}{8} X^2 - X Xb - \frac{1}{8} Xb^2 + X - Xb + 1 \\
 & - \frac{625}{9216} X^6, + \dots +, - \frac{1}{12} X^3 - \frac{5}{4} X Xb^2 + \frac{4}{3} Xb^3 - \frac{1}{8} X^2 - X Xb + \frac{9}{8} Xb^2 - X + Xb + 1
 \end{aligned}$$

Procédures utiles pour calculer une fonction perturbatrice planétaire (méthode de Laplace, calculs par inégalité)

- > read cat(chemin, 'initmo.pr'):
- > read cat(chemin, 'IngUind.pr'):
- > read cat(chemin, 'IngUDLap.pr'):
- > read cat(chemin, 'EqLag.pr');
- > alias(F=hypergeom):

Dev3.4.4

## 2. Perturbations à très longues périodes des orbites des planètes

On construit l'inégalité séculaire (termes indépendants des longitudes moyennes) dans la fonction perturbatrice d'une planète perturbée par une autre, au degré 2 en excentricités et inclinaisons.

### 2.1. Système formel de Laplace Lagrange à l'ordre 1 des masses

Pour obtenir le système linéaire de Laplace Lagrange, il faut calculer la fonction perturbatrice au degré 2 :

```
> Dmin:=0; Dmax:=2;
```

$$Dmin := 0$$

$$Dmax := 2$$

Initialisation des développements du mouvement képlérien au degré choisi

```
> deb0:=time():
```

```
> inineg:=IniINEGLap(Dmin,Dmax):
```

```
> XnmT:= op(1,inineg):
```

```
> COSSYm1T:=TRONC(COSSYm1,Dmax,[Y[i],Yb[i],Y[j],Yb[j]]):
```

```
> for k from 0 to Dmax/2 do: DELTACOSSYmT.k:=op(k+2,inineg) od:
```

```
> s:='s': k:='k': m:='m':
```

```
> Usn:=expand(sum(sum(Ca(s,k,m,alpha)*beta^(k)*(beta^2-1)^m*W^k,k=0..Dma
```

```
> x/2),m=0..Dmax)):
```

```
> DbetaWn:=combine(Usn,power): nops(%);
```

```
> print(time()-deb0):
```

12

.019

On choisit l'inégalité séculaire :

```
> pp:=0; qq:=0;
```

$$pp := 0$$

$$qq := 0$$

Calcul de la partie indirecte concernant cette inégalité (on vérifie en fait ici qu'elle est nulle):

```
> ingINDi:=INEG(pp,qq,Dmin,Dmax,1,-2,COSSYm1T,XnmT):
```

```
> PRINTRONC(ingINDi,2,varxyij):
```

```
> ingINDj:=INEG(pp,qq,Dmin,Dmax,-2,1,COSSYm1T,XnmT):
```

```
> PRINTRONC(ingINDj,2,varxyij):
```

```
> UiPerturBParj:=alpha*ingINDi:
```

```
> UjPerturBParj:=1/alpha^2*ingINDj:
```

$$0$$

$$0$$

Calcul de la partie directe en coefficients de Laplace :

```
> UD1:=INVDLAP(pp,qq,Dmin,2,DbetaWn,DELTACOSSYmT,XnmT):
```

```
> UD1:=sort(collect(UD1,varxyij,distributed),varxyij): op(1,UD1):
```

$$(2 \operatorname{Ca}(0, 0, 2, \alpha) + \frac{3}{2} \operatorname{Ca}(0, 0, 1, \alpha)) x_i x_{b_i}$$

```
> UDa:=eval(subs(Ca=Calpha,UD1)): op(1,%):
```

$$\left(\frac{9}{32} \alpha^4 F\left(\left[\frac{5}{2}, \frac{5}{2}\right], [3], \alpha^2\right) + \frac{3}{8} \alpha^2 F\left(\left[\frac{3}{2}, \frac{3}{2}\right], [2], \alpha^2\right)\right) x_i x_{b_i}$$

```
> UDai:=sort(collect(UDa-UiPerturBParj,varxyij,distributed),varxyij):
```

```
> UPa[i]:=K*m[j]/a[j]*UDai*eL[i]^pp*eL[j]^qq:
```

```
> UDaj:=sort(collect(UDa-UjPerturBParj,varxyij,distributed),varxyij):
```

> UPa[j]:=K\*m[i]/a[j]\*UDaj\*eL[i]^pp\*eL[j]^qq;

Fonction perturbatrice puis équations de Lagrange pour une planète **intérieure** (indice *i*) perturbée par une planète **extérieure** (indice *j*)

> Up:=TRONC(UPa[i],Dmax,varxyij); nops(%);

$$\begin{aligned}
 U_p := & \frac{K m_j}{a_j} \left( \left( \frac{9}{32} \alpha^4 F\left(\left[\frac{5}{2}, \frac{5}{2}\right], [3], \alpha^2\right) + \frac{3}{8} \alpha^2 F\left(\left[\frac{3}{2}, \frac{3}{2}\right], [2], \alpha^2\right) \right) x_i x_{b_i} + \left( -\frac{15}{64} \alpha^5 F\left(\left[\frac{5}{2}, \frac{7}{2}\right], [4], \alpha^2\right) - \frac{15}{32} \alpha^3 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [3], \alpha^2\right) \right) x_i x_{b_j} \right. \\
 & + \left( -\frac{15}{64} \alpha^5 F\left(\left[\frac{5}{2}, \frac{7}{2}\right], [4], \alpha^2\right) - \frac{15}{32} \alpha^3 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [3], \alpha^2\right) \right) x_{b_i} x_j + \left( \frac{9}{32} \alpha^4 F\left(\left[\frac{5}{2}, \frac{5}{2}\right], [3], \alpha^2\right) + \frac{3}{8} \alpha^2 F\left(\left[\frac{3}{2}, \frac{3}{2}\right], [2], \alpha^2\right) \right) x_j x_{b_j} \\
 & + \left( 2 C_i C_j \left( \frac{3}{4} \alpha^2 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha^2\right) + \frac{525}{256} \alpha^6 F\left(\left[\frac{7}{2}, \frac{9}{2}\right], [4], \alpha^2\right) - \frac{45}{32} \alpha^4 F\left(\left[\frac{5}{2}, \frac{7}{2}\right], [3], \alpha^2\right) \right) \right. \\
 & + \left. 2 C_i C_j \left( -\frac{525}{128} \alpha^6 F\left(\left[\frac{7}{2}, \frac{9}{2}\right], [4], \alpha^2\right) + \frac{45}{32} \alpha^4 F\left(\left[\frac{5}{2}, \frac{7}{2}\right], [3], \alpha^2\right) \right) + \frac{525}{128} C_i C_j \alpha^6 F\left(\left[\frac{7}{2}, \frac{9}{2}\right], [4], \alpha^2\right) \right) y_i y_{b_j} \\
 & + \left( 2 C_i C_j \left( \frac{3}{4} \alpha^2 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha^2\right) + \frac{525}{256} \alpha^6 F\left(\left[\frac{7}{2}, \frac{9}{2}\right], [4], \alpha^2\right) - \frac{45}{32} \alpha^4 F\left(\left[\frac{5}{2}, \frac{7}{2}\right], [3], \alpha^2\right) \right) \right. \\
 & + \left. 2 C_i C_j \left( -\frac{525}{128} \alpha^6 F\left(\left[\frac{7}{2}, \frac{9}{2}\right], [4], \alpha^2\right) + \frac{45}{32} \alpha^4 F\left(\left[\frac{5}{2}, \frac{7}{2}\right], [3], \alpha^2\right) \right) + \frac{525}{128} C_i C_j \alpha^6 F\left(\left[\frac{7}{2}, \frac{9}{2}\right], [4], \alpha^2\right) \right) y_{b_i} y_j \\
 & - \frac{3}{2} \alpha^2 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha^2\right) y_i y_{b_i} - \frac{3}{2} \alpha^2 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha^2\right) y_j y_{b_j} + F\left(\left[\frac{1}{2}, \frac{1}{2}\right], [1], \alpha^2\right)
 \end{aligned}$$

4

> DVai:=EqLag(Up,"interieur",i,j): seq(nops(DVai[k]),k=1..4);  
1, 9, 2, 2

Fonction perturbatrice puis équations de Lagrange pour une planète **extérieure** (indice *j*) perturbée par une planète **intérieure** (indice *i*)

> Uops:=TRONC(UPa[j],Dmax,varxyij); nops(%);

$$\begin{aligned}
 Up := & \frac{K m_i}{a_j} \left( \left( \frac{9}{32} \alpha^4 F\left(\frac{5}{2}, \frac{5}{2}, [3], \alpha^2\right) + \frac{3}{8} \alpha^2 F\left(\frac{3}{2}, \frac{3}{2}, [2], \alpha^2\right) \right) x_i x b_i + \left( -\frac{15}{64} \alpha^5 F\left(\frac{5}{2}, \frac{7}{2}, [4], \alpha^2\right) - \frac{15}{32} \alpha^3 F\left(\frac{3}{2}, \frac{5}{2}, [3], \alpha^2\right) \right) x_i x b_j \right. \\
 & + \left( -\frac{15}{64} \alpha^5 F\left(\frac{5}{2}, \frac{7}{2}, [4], \alpha^2\right) - \frac{15}{32} \alpha^3 F\left(\frac{3}{2}, \frac{5}{2}, [3], \alpha^2\right) \right) x b_i x_j + \left( \frac{9}{32} \alpha^4 F\left(\frac{5}{2}, \frac{5}{2}, [3], \alpha^2\right) + \frac{3}{8} \alpha^2 F\left(\frac{3}{2}, \frac{3}{2}, [2], \alpha^2\right) \right) x_j x b_j \\
 & + \left( 2 C_i C_j \left( \frac{3}{4} \alpha^2 F\left(\frac{3}{2}, \frac{5}{2}, [2], \alpha^2\right) + \frac{525}{256} \alpha^6 F\left(\frac{7}{2}, \frac{9}{2}, [4], \alpha^2\right) - \frac{45}{32} \alpha^4 F\left(\frac{5}{2}, \frac{7}{2}, [3], \alpha^2\right) \right) \right. \\
 & + \left. 2 C_i C_j \left( -\frac{525}{128} \alpha^6 F\left(\frac{7}{2}, \frac{9}{2}, [4], \alpha^2\right) + \frac{45}{32} \alpha^4 F\left(\frac{5}{2}, \frac{7}{2}, [3], \alpha^2\right) \right) + \frac{525}{128} C_i C_j \alpha^6 F\left(\frac{7}{2}, \frac{9}{2}, [4], \alpha^2\right) \right) y_i y b_j \\
 & + \left( 2 C_i C_j \left( \frac{3}{4} \alpha^2 F\left(\frac{3}{2}, \frac{5}{2}, [2], \alpha^2\right) + \frac{525}{256} \alpha^6 F\left(\frac{7}{2}, \frac{9}{2}, [4], \alpha^2\right) - \frac{45}{32} \alpha^4 F\left(\frac{5}{2}, \frac{7}{2}, [3], \alpha^2\right) \right) \right. \\
 & + \left. 2 C_i C_j \left( -\frac{525}{128} \alpha^6 F\left(\frac{7}{2}, \frac{9}{2}, [4], \alpha^2\right) + \frac{45}{32} \alpha^4 F\left(\frac{5}{2}, \frac{7}{2}, [3], \alpha^2\right) \right) + \frac{525}{128} C_i C_j \alpha^6 F\left(\frac{7}{2}, \frac{9}{2}, [4], \alpha^2\right) \right) y b_i y_j \\
 & - \frac{3}{2} \alpha^2 F\left(\frac{3}{2}, \frac{5}{2}, [2], \alpha^2\right) y_i y b_i - \frac{3}{2} \alpha^2 F\left(\frac{3}{2}, \frac{5}{2}, [2], \alpha^2\right) y_j y b_j + F\left(\frac{1}{2}, \frac{1}{2}, [1], \alpha^2\right)
 \end{aligned}$$

4

```

> DVaj:=EqLag(Up, "exterieur", i, j):      seq(nops(DVaj[k]), k=1..4);
1, 10, 2, 2
    
```

Affichage des équations en excentricité et inclinaison du système de Laplace Lagrange:

```

> for kv from 3 to 4 do:
> coeff(coeff(DVai[kv], eL[i], pp), eL[j], qq):
> TRONC(%, 1, varxyij):
> sort(collect(%, varxyij, distributed, factor), varxyij):
> subs(alpha=alpha[i, j], K=n[i]^2*a[i]^3/(M+m[i]), a[i]=alpha[i, j]*a[j],
> %):
> if kv=3 then: Dx[i]:=?:print('Dx[i]=', Dx[i]) else:
> Dy[i]:=?:print('Dy[i]=', Dy[i]) fi:
    
```



```

> coeff(coeff(DVaj[kv],eL[i],pp),eL[j],qq):
> TRONC(% ,1,varxyij):
> sort(collect(% ,varxyij,distributed,factor),varxyij):
> subs(alpha=alpha[i,j],K=n[j]^2*a[j]^3/(M+m[j]),a[i]=alpha[i,j]*a[j],
> %):
> if kv=3 then: Dx[j]:=%:print('Dx[j]=',Dx[j]) else:
> Dy[j]:=%:print('Dy[j]=',Dy[j]) fi:
> od:

```

$$Dx[i] = \frac{3}{16} \frac{I \phi_i n_i \alpha_{i,j}^3 m_j (3 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2) + 4 F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{i,j}^2)) x_i}{M + m_i}$$

$$- \frac{15}{32} \frac{I \phi_i n_i \alpha_{i,j}^4 m_j (\alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2) + 2 F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2)) x_j}{M + m_i}$$

$$Dx[j] = - \frac{15}{32} \frac{I \phi_j n_j m_i \alpha_{i,j}^3 (\alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2) + 2 F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2)) x_i}{M + m_j}$$

$$+ \frac{3}{16} \frac{I \phi_j n_j m_i \alpha_{i,j}^2 (3 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2) + 4 F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{i,j}^2)) x_j}{M + m_j}$$

$$Dy[i] = - \frac{3}{4} \frac{I n_i \alpha_{i,j}^3 m_j F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2) y_i}{\phi_i (M + m_i)} + \frac{3}{4} \frac{I n_i \alpha_{i,j}^3 m_j C_i C_j F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2) y_j}{\phi_i (M + m_i)}$$

$$Dy[j] = \frac{3}{4} \frac{I n_j m_i C_i C_j \alpha_{i,j}^2 F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2) y_i}{\phi_j (M + m_j)} - \frac{3}{4} \frac{I n_j m_i \alpha_{i,j}^2 F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2) y_j}{\phi_j (M + m_j)}$$

substitution des fonctions  $\phi_i, \phi_j, C_i$  et  $C_j$  par leur développement :

```
> subs(phi[i]=sqrt(1-x[i]*xb[i]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dx[i]):
> Dx[i]:=mtaylor(%,varxyij,2);
> subs(phi[j]=sqrt(1-x[j]*xb[j]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dx[j]):
> Dx[j]:=mtaylor(%,varxyij,2);
> subs(phi[i]=sqrt(1-x[i]*xb[i]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dy[i]):
> Dy[i]:=mtaylor(%,varxyij,2);
> subs(phi[j]=sqrt(1-x[j]*xb[j]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dy[j]):
> Dy[j]:=mtaylor(%,varxyij,2);
```

$$Dx_i := \frac{3}{16} \frac{In_i \alpha_{i,j}^3 m_j (3 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2) + 4 F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{i,j}^2)) x_i}{M + m_i}$$

$$- \frac{15}{32} \frac{In_i \alpha_{i,j}^4 m_j (\alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2) + 2 F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2)) x_j}{M + m_i}$$

$$Dx_j := - \frac{15}{32} \frac{In_j m_i \alpha_{i,j}^3 (\alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2) + 2 F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2)) x_i}{M + m_j}$$

$$+ \frac{3}{16} \frac{In_j m_i \alpha_{i,j}^2 (3 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2) + 4 F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{i,j}^2)) x_j}{M + m_j}$$

$$Dy_i := -\frac{3}{4} \frac{I n_i \alpha_{i,j}^3 m_j F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha_{i,j}^2\right) y_i}{M + m_i} + \frac{3}{4} \frac{I n_i \alpha_{i,j}^3 m_j F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha_{i,j}^2\right) y_j}{M + m_i}$$

$$Dy_j := \frac{3}{4} \frac{I n_j m_i \alpha_{i,j}^2 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha_{i,j}^2\right) y_i}{M + m_j} - \frac{3}{4} \frac{I n_j m_i \alpha_{i,j}^2 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha_{i,j}^2\right) y_j}{M + m_j}$$

On prépare également ici le calcul des perturbations séculaires de la longitude moyenne, limitées au degré 2 en excentricités et inclinaisons :

```
> varxij:=[x[i],xb[i],x[j],xb[j]]:varyij:=[y[i],yb[i],y[j],yb[j]]:
> TRONC(DVai[2],2,varxyij):
> sort(collect(%,varxyij,distributed,factor),varxyij):
> subs(alpha=alpha[i,j],K=n[i]^2*a[i]^3/(M+m[i]),a[i]=alpha[i,j]*a[j],
> %):
> Dq[i]:=:%: #print('Dq[i]=',Dq[i]):
> subs(phi[i]=sqrt(1-x[i]*xb[i]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dq[i]):
> Dq[i]:=mtaylor(%,varxyij,3):
```

L'équation  $dq_i/dt$  réduite à sa partie séculaire ne comporte que des termes pairs en excentricités et inclinaisons et pairs en inclinaisons (caractéristique nulle); pour ne pas surcharger l'affichage on sépare les termes par degrés; le degré 0 vaut:

```
> TRONCMINMAX(Dq[i],0,0,varxyij);
```

$$-\frac{n_i \alpha_{i,j}^3 m_j F\left(\left[\frac{3}{2}, \frac{3}{2}\right], [2], \alpha_{i,j}^2\right)}{M + m_i}$$

partie de degré 2 en excentricités :

> TRONCMINMAX(Dq[i], 2, 2, varxij) ;

$$\begin{aligned} & \frac{3}{64} \frac{n_i \alpha_{i,j}^2 m_j (24 F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{i,j}^2) \alpha_{i,j} a_j + 78 \alpha_{i,j}^3 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2) a_j + 50 \alpha_{i,j}^5 F([\frac{7}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2) a_j) x_i x b_i}{a_j (M + m_i)} \\ & + \frac{15}{512} \frac{\alpha_{i,j}^3 n_i m_j (88 F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2) \alpha_{i,j} a_j + 70 \alpha_{i,j}^5 F([\frac{7}{2}, \frac{9}{2}], [5], \alpha_{i,j}^2) a_j + 156 \alpha_{i,j}^3 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2) a_j) x_i x b_j}{a_j (M + m_i)} \\ & + \frac{15}{512} \frac{\alpha_{i,j}^3 n_i m_j (88 F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2) \alpha_{i,j} a_j + 70 \alpha_{i,j}^5 F([\frac{7}{2}, \frac{9}{2}], [5], \alpha_{i,j}^2) a_j + 156 \alpha_{i,j}^3 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2) a_j) x b_i x_j}{a_j (M + m_i)} \\ & - \frac{3}{32} \frac{n_i \alpha_{i,j}^3 m_j (16 F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{i,j}^2) + 25 \alpha_{i,j}^4 F([\frac{7}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2) + 42 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2)) x_j x b_j}{M + m_i} \end{aligned}$$

partie de degré 2 en inclinaisons :

> TRONCMINMAX(Dq[i], 2, 2, varyij) ;

$$\begin{aligned} & \frac{3}{4} \frac{n_i \alpha_{i,j}^2 m_j (6 \alpha_{i,j} F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2) a_j + 15 \alpha_{i,j}^3 F([\frac{5}{2}, \frac{7}{2}], [3], \alpha_{i,j}^2) a_j) y_i y b_i}{a_j (M + m_i)} \\ & - \frac{3}{4} \frac{n_i \alpha_{i,j}^2 m_j (15 \alpha_{i,j}^3 F([\frac{5}{2}, \frac{7}{2}], [3], \alpha_{i,j}^2) a_j + 7 \alpha_{i,j} F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2) a_j) y_i y b_j}{a_j (M + m_i)} \\ & - \frac{3}{4} \frac{n_i \alpha_{i,j}^2 m_j (15 \alpha_{i,j}^3 F([\frac{5}{2}, \frac{7}{2}], [3], \alpha_{i,j}^2) a_j + 7 \alpha_{i,j} F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2) a_j) y b_i y_j}{a_j (M + m_i)} \\ & + \frac{3}{4} \frac{n_i \alpha_{i,j}^3 m_j (15 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [3], \alpha_{i,j}^2) + 8 F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2)) y_j y b_j}{M + m_i} \end{aligned}$$

Pour la planète extérieure (indice j) on a de même :

- > TRONC(DVaj[2], 2, varxyij) :
  - > sort(collect(% , varxyij, distributed, factor), varxyij) :
  - > subs(alpha=alpha[i, j], K=n[j]^2\*a[j]^3/(M+m[j]), a[i]=alpha[i, j]\*a[j], %):
  - > Dq[j]:= %: #print('Dq[j]=', Dq[j]) :
  - > subs(phi[j]=sqrt(1-x[j]\*xb[j]), C[i]=sqrt(1-y[i]\*yb[i]), %):
  - > C[j]=sqrt(1-y[j]\*yb[j]), Dq[j] :
  - > Dq[j]:=mtaylor(% , varxyij, 3) : TRONCMINMAX(Dq[j], 0, 0, varxyij) ;
- $$\frac{n_j m_i (\alpha_{i,j}^2 F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{i,j}^2) + 2 F([\frac{1}{2}, \frac{1}{2}], [1], \alpha_{i,j}^2))}{M + m_j}$$
- > TRONCMINMAX(Dq[j], 2, 2, varxi j) ;

$$\begin{aligned}
 & \frac{3}{32} \frac{\alpha_{i,j}^2 n_j m_i (24 F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{i,j}^2) + 48 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2) + 25 \alpha_{i,j}^4 F([\frac{7}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2))}{M + m_j} x_i x b_i \\
 & - \frac{15}{512} \frac{\alpha_{i,j}^3 n_j m_i (70 \alpha_{i,j}^4 F([\frac{7}{2}, \frac{9}{2}], [5], \alpha_{i,j}^2) + 136 F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2) + 180 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2))}{M + m_j} x_i x b_j \\
 & - \frac{15}{512} \frac{\alpha_{i,j}^3 n_j m_i (70 \alpha_{i,j}^4 F([\frac{7}{2}, \frac{9}{2}], [5], \alpha_{i,j}^2) + 136 F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2) + 180 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2))}{M + m_j} x b_i x_j \\
 & + \frac{3}{64} \frac{\alpha_{i,j}^2 n_j m_i (56 F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{i,j}^2) + 50 \alpha_{i,j}^4 F([\frac{7}{2}, \frac{7}{2}], [4], \alpha_{i,j}^2) + 102 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{i,j}^2))}{M + m_j} x_j x b_j
 \end{aligned}$$

> TRONCMINMAX(Dq[j], 2, 2, varyij);

$$\begin{aligned}
 & - \frac{9}{4} \frac{\alpha_{i,j}^2 n_j m_i (5 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [3], \alpha_{i,j}^2) + 4 F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2))}{M + m_j} y_i y b_i \\
 & + \frac{3}{4} \frac{\alpha_{i,j}^2 n_j m_i (15 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [3], \alpha_{i,j}^2) + 13 F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2))}{M + m_j} y_i y b_j \\
 & + \frac{3}{4} \frac{\alpha_{i,j}^2 n_j m_i (15 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [3], \alpha_{i,j}^2) + 13 F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2))}{M + m_j} y b_i y_j \\
 & - \frac{3}{4} \frac{\alpha_{i,j}^2 n_j m_i (14 F([\frac{3}{2}, \frac{5}{2}], [2], \alpha_{i,j}^2) + 15 \alpha_{i,j}^2 F([\frac{5}{2}, \frac{7}{2}], [3], \alpha_{i,j}^2))}{M + m_j} y_j y b_j
 \end{aligned}$$

> dVsecxyij := [Dq[i], Dx[i], Dy[i], Dq[j], Dx[j], Dy[j]]:

> save dVsecxyij , cat(chemin, 'dVsecxyij'.Dmax.';m'):

## 2.2. Application aux planètes du système solaire

Pour limiter l'affichage des développements, on ne considère ici que 4 planètes, mais on pourrait bien sûr étendre les résultats suivants à tout autre nombre de planètes.

```
> npl:=4;
                                     npl := 4
> with(linalg):
Warning, new definition for norm
Warning, new definition for trace
```

On remplace  $i$  et  $j$  par des valeurs entre 1 et 4 dans l'expression formelle des perturbations séculaires obtenues pour  $i$  et  $j$  quelconques:

```
> Ax:='Ax': Ax:=array(1..npl): Ay:='Ay': Ay:=array(1..npl): Aq:='Aq':
> Aq:=array(1..npl):
> for k1 to npl do: Ax[k1]:=0: Ay[k1]:=0: Aq[k1]:=0: od:
> varxy18:=[seq(x[k],k=1..npl),seq(xb[k],k=1..npl),seq(y[k],k=1..npl),s
> eq(yb[k],k=1..npl)];
                                     varxy18 := [x1, x2, x3, x4, xb1, xb2, xb3, xb4, y1, y2, y3, y4, yb1, yb2, yb3, yb4]
> for k1 from 1 to npl do:
> for k2 from 1 to npl do:
> if k1<>k2 then:
> if k1<k2 then Ax[k1]:=Ax[k1]+subs(i=k1,j=k2,Dx[i]) elif k1>k2
> then Ax[k1]:=Ax[k1]+subs(i=k2,j=k1,Dx[j]) fi:
> if k1<k2 then Ay[k1]:=Ay[k1]+subs(i=k1,j=k2,Dy[i]) elif k1>k2
> then Ay[k1]:=Ay[k1]+subs(i=k2,j=k1,Dy[j]) fi:
```

```
> if k1<k2 then Aq[k1]:=Aq[k1]+subs(i=k1,j=k2,Dq[i]) elif k1>k2
> then Aq[k1]:=Aq[k1]+subs(i=k2,j=k1,Dq[j]) fi:
> fi:
> od: od:
```

Voici par exemple l'équation  $dx_2/dt$  du système en excentricités relative à la planète numero 2 (équation linéaire par rapport aux  $x_i$ , où l'on voit bien où interviennent les masses perturbatrices  $m_1, m_3$  et  $m_4$ ) :

```
> print(collect(Ax[2],varxy18));
```

$$\begin{aligned}
 & -\frac{15}{32} \frac{I n_2 m_1 \alpha_{1,2}^3 (\alpha_{1,2}^2 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{1,2}^2) + 2F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{1,2}^2))}{M + m_2} x_1 \\
 & + \left( \frac{3}{16} \frac{I n_2 m_1 \alpha_{1,2}^2 (3 \alpha_{1,2}^2 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{1,2}^2) + 4F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{1,2}^2))}{M + m_2} \right. \\
 & + \frac{3}{16} \frac{I n_2 \alpha_{2,3}^3 m_3 (3 \alpha_{2,3}^2 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{2,3}^2) + 4F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{2,3}^2))}{M + m_2} \\
 & \left. + \frac{3}{16} \frac{I n_2 \alpha_{2,4}^3 m_4 (3 \alpha_{2,4}^2 F([\frac{5}{2}, \frac{5}{2}], [3], \alpha_{2,4}^2) + 4F([\frac{3}{2}, \frac{3}{2}], [2], \alpha_{2,4}^2))}{M + m_2} \right) x_2 \\
 & - \frac{15}{32} \frac{I n_2 \alpha_{2,4}^4 m_4 (\alpha_{2,4}^2 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{2,4}^2) + 2F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{2,4}^2))}{M + m_2} x_4 \\
 & - \frac{15}{32} \frac{I n_2 \alpha_{2,3}^4 m_3 (\alpha_{2,3}^2 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_{2,3}^2) + 2F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_{2,3}^2))}{M + m_2} x_3
 \end{aligned}$$



de même pour la partie de degré 0 de l'équation  $dq_2/dt$ , et l'équation en inclinaisons  $dy_2/dt$  :

```
> TRONC(Aq[2], 0, varxy18);
```

$$-\frac{n_2 \alpha_2, 3^3 m_3 F\left(\left[\frac{3}{2}, \frac{3}{2}\right], [2], \alpha_2, 3^2\right)}{M + m_2} + \frac{n_2 m_1 (\alpha_1, 2^2 F\left(\left[\frac{3}{2}, \frac{3}{2}\right], [2], \alpha_1, 2^2\right) + 2 F\left(\left[\frac{1}{2}, \frac{1}{2}\right], [1], \alpha_1, 2^2\right))}{M + m_2} - \frac{n_2 \alpha_2, 4^3 m_4 F\left(\left[\frac{3}{2}, \frac{3}{2}\right], [2], \alpha_2, 4^2\right)}{M + m_2}$$

```
> print(collect(Ay[2], varxy18));
```

$$\left( -\frac{3}{4} \frac{I n_2 m_1 \alpha_1, 2^2 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha_1, 2^2\right)}{M + m_2} - \frac{3}{4} \frac{I n_2 \alpha_2, 3^3 m_3 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha_2, 3^2\right)}{M + m_2} - \frac{3}{4} \frac{I n_2 \alpha_2, 4^3 m_4 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha_2, 4^2\right)}{M + m_2} \right) y_2$$

$$+ \frac{3}{4} \frac{I n_2 m_1 \alpha_1, 2^2 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha_1, 2^2\right) y_1}{M + m_2} + \frac{3}{4} \frac{I n_2 \alpha_2, 4^3 m_4 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha_2, 4^2\right) y_4}{M + m_2} + \frac{3}{4} \frac{I n_2 \alpha_2, 3^3 m_3 F\left(\left[\frac{3}{2}, \frac{5}{2}\right], [2], \alpha_2, 3^2\right) y_3}{M + m_2}$$

On tire de ces développements les éléments des matrices du système de Laplace -Lagrange :

```
> LapLagx:=array(1..npl,1..npl): LapLagy:=array(1..npl,1..npl):
> for k1 to npl do:
> for k2 to npl do:
> LapLagx[k1,k2]:=coeff(Ax[k1],x[k2],1):
> LapLagy[k1,k2]:=coeff(Ay[k1],y[k2],1):
> od:
> od:
```

Par exemple, terme de la matrice en excentricité en facteur de  $x_1$  (colonne 1) dans l'équation  $dx_2/dt$  (ligne 2):

```
> print(LapLagx[2,1]);
```

$$\frac{15}{32} \frac{I n_2 m_1 \alpha_1, 2^3 (\alpha_1, 2^2 F([\frac{5}{2}, \frac{7}{2}], [4], \alpha_1, 2^2) + 2 F([\frac{3}{2}, \frac{5}{2}], [3], \alpha_1, 2^2))}{M + m_2}$$

## Données relatives aux 8 grosses planètes constantes Bretagnon 1982 (VSOP82)

date correspondant a ces elements:

J2000 (2451545.0)

Valeurs des variables x et y (éléments moyens) à cette date:

0.04466059760d0,-0.00449282133d0,-0.00374081650d0, 0.08536560252d0, 0.04698572124d0,-0.00296003595d0,-0.04595132376d0, 0.00599977571d0,  
 0.20072331368d0, 0.00506684726d0, 0.01628447663d0,-0.03789973236d0, 0.01200385748d0, 0.05542964254d0, 0.00563791307d0, 0.00669242413d0,  
 0.04061563384d0, 0.00682410142d0, 0.00000000000d0 , 0.01047042574d0,-0.00206561098d0,-0.00871747436d0, 0.00185915075d0,-0.01029147819d0,  
 0.04563550461d0, 0.02882285775d0, 0.00000000000d0 , 0.01228449307d0, 0.01118377157d0, 0.01989147301d0, 0.00648617008d0, 0.01151683985d0,

Valeurs des longitudes moyennes à cette date:

0.4402593038E+01 0.3176150173E+01 0.1753532299E+01 0.6203423760E+01 0.599520598E+00 0.874060889E+00 0.5481206516E+01 0.531260771E+01

moyens mouvements moyens (rad/an)

26.0879031415742D0, 10.2132855462110D0, 6.2830758491800D0, 3.3406124314923D0,  
 0.5296909650946D0, 0.2132990954380D0, 0.0747815985673D0, 0.0381330356378D0,

inverses des masses

6023600.D0, 408523.5D0, 328900.5D0, 3098710.D0, 1047.355D0,3498.5D0,22869D0,19314.D0,

constantes p0i

0.11795121D-5, 0.49815442D-5,-0.15429922D-5, 1.20304680D-5,  
 5.70597419D-5,-249.58418298D-5, -271.86256199D-5,-271.78052050D-5,

Constante de Gauss GK

0.1720209895000000D-01

Calcul des constantes  $A(i)$  en fonction des moyens mouvements moyens:

```
> GK1:=(0.01720209895000000*365.25)^2:
```

Donc, implicitement avec cette constante, l'unité de temps est l'année, et les moyens mouvements moyens suivants sont donnés en radians par an.

```
> N[1]:=26.0879031415742: N[2]:=10.2132855462110:
> N[3]:=6.2830758491800: N[4]:=3.3406124314923: N[5]:=0.5296909650946:
> N[6]:=0.2132990954380: N[7]:=0.0747815985673:
> N[8]:=0.0381330356378:
> TAM[1]:=6023600.0: TAM[2]:=408523.5: TAM[3]:=328900.5:
> TAM[4]:=3098710.0: TAM[5]:=1047.355: TAM[6]:=3498.5: TAM[7]:=22869.0:
> TAM[8]:=19314.0:
> Poi[1]:=0.11795121: Poi[2]:=0.49815442: Poi[3]:=-0.15429922:
> Poi[4]:=1.20304680: Poi[5]:=5.70597419: Poi[6]:=-249.58418298:
> Poi[7]:=-271.86256199: Poi[8]:=-271.78052050:
> for k to 8 do:
> TMAS[k]:=evalf(1/TAM[k],15):Po[k]:=evalf(Poi[k]*0.00001,15):
> AAM[k]:=evalf(N[k]*(1+Po[k]),15):
> A[k]:=evalf((GK1/AAM[k]^2*(1+TMAS[k]))^(1/3),15):
> od:
> print(A):
```

```
table([
4 = 1.52367945741418
5 = 5.20260715684208
6 = 9.55476474198935
7 = 19.2181895403685
8 = 30.1098766317019
1 = .387098371564837
2 = .723329826016509
3 = 1.00000106514317
])
```

```

> for k1 to 7 do :
> for k2 from k1+1 to 8 do:
> alf[k1,k2]:=evalf(A[k1]/A[k2],15):
> alf[k2,k1]:=alf[k1,k2]:
> od:
> od:

> save npl,TMAS,N,A,AAM,alf, cat(chemin,'init_pla.txt'):

```

### 2.3. Solution numérique du système de Laplace-Lagrange

C<sub>26.2.3</sub>

Evaluation numérique du système en excentricités. On ne considère ici que les 4 planètes extérieures Jupiter, Saturne, Uranus et Neptune, d'indices 5 à 8 dans le paragraphe précédent. Pour manipuler les matrices (4x4) correspondant à  $npl=4$ , on introduit un index ( $dk=4$ ) pour ramener les indices 5 à 8 aux valeurs 1 à 4:

```

> XX:=matrix(npl,npl):VX:=vector(npl): dk:=4;
                                dk := 4
> for k to npl do:
> for l to npl do:
> XX[k,l]:= evalf(subs(M=1,seq(m[k1]=TMAS[k1+dk],k1=1..npl),
> seq(n[k1]=AAM[k1+dk],k1=1..npl),seq(seq(alpha[k1,k2]=alf[k1+dk,k2+dk],
> k2=1..npl),k1=1..npl), LapLagx[k,l]),20):
> od: od:
> print(XX);

```

$$\begin{bmatrix} .000036087392426113614299 I & -.000023223299875333125335 I & -.13315983007027951571 10^{-6} I & -.24145612534769323460 10^{-7} I \\ -.000057260823076696290321 I & .000089509729979805207444 I & -.87949904032333993736 10^{-6} I & -.12614083808229416486 10^{-6} I \\ -.15134863640144569934 10^{-5} I & -.40542170326651510225 10^{-5} I & .000013293559737527025275 I & -.15233714904161419588 10^{-5} I \\ -.18516911907735374746 10^{-6} I & -.39233078673041774009 10^{-6} I & -.10278526896024212083 10^{-5} I & .32313574720928787439 10^{-5} I \end{bmatrix}$$

Maple ne résoud pas directement un système de la forme :  $\frac{dX}{dt} = A X$  où  $A$  est une matrice carrée numérique.

Il faut donc exprimer la solution connue :  $X = P e^{(D t)} P^{-1} X_0$  où  $D$  est la matrice diagonale des valeurs propres de  $A$ ,  $P$  la matrice des vecteurs propres,  $\text{inv}(P)$  son inverse et  $X_0$  la valeur de  $X$  à l'instant initial.

Calcul des valeurs propres de la matrice :

```
> Digits:=20;
```

```
Digits := 20
```

```
> lambda[exc]:= [eigenvalues(XX)];
```

```
lambda_exc := [.00010802449768716235322 I, .000017935389575491109393 I, .000013091997026140007792 I, .30701553267452553478 10^{-5} I]
```

périodes correspondantes , en années :

```
> for k to npl do:
> print(evalf(2*Pi/lambda[exc][k]*I,10));
> od;
```

```
58164.44828
350323.3247
479925.6594
.2046536621 10^7
```

Calcul des vecteurs propres :

```
> vx:=[ eigenvectors(XX) ];
```

```
vx := [[.00010802449768716235322 I, 1, {[.30695755331641364419, -.95104758982733399253, .035755094936814944003, .0026674953944675523185]}],
[.000017935389575491109393 I, 1, {[.63940413482000910381 I, .50351431815574310935 I, -.65625778169377757480 I, .024387477836945310795 I]}],
[.000013091997026140007792 I, 1, {[.08527962873397042690 I, .07763094906832026625 I, 1.2117395561483676881 I, -.13099940626113734020 I]}],
[.30701553267452553478 10-5 I, 1, {[.006433718018462100846 I, .007248583957490107807 I, .15144779143659636351 I, .99068897602237082192 I]}]]
```

Pour  $npl = 4$ , on obtient ainsi un tableau  $vx[4][3]$  tel que  $vx[k][1]$  contient la  $k$  ième valeur propre,  $vx[k][2]$  sa multiplicité  $m$  et  $vx[k][3]$  un ensemble de  $m$  vecteurs propres constitués chacun de 4 éléments.

Pour reconstituer la matrice des vecteurs propres, c'est assez compliqué; analysons le premier vecteur propre, associé à la première valeur propre :

```
> vx[1][3];nops(%);
{[.30695755331641364419, -.95104758982733399253, .035755094936814944003, .0026674953944675523185]}
1
```

```
> op(1,vx[1][3]);nops(%);
[.30695755331641364419, -.95104758982733399253, .035755094936814944003, .0026674953944675523185]
3
```

```
> op(3,op(1,vx[1][3]));nops(%);
[4 = .0026674953944675523185, 1 = .30695755331641364419, 2 = -.95104758982733399253, 3 = .035755094936814944003]
4
```

analysons le deuxième élément :

```
> op(2,op(3,op(1,vx[1][3])));nops(%);
1 = .30695755331641364419
2
```

```
> op(1,op(2,op(3,op(1,vx[1][3]))));
> op(2,op(2,op(3,op(1,vx[1][3]))));
```

1  
.30695755331641364419

Avec une multiplicité 1, le vecteur propre associé à la valeur propre  $k$ , est donné par:  $V = \text{op}(3, \text{op}(1, \text{vx}[k][3]))$  et pour  $h$  entre 1 et 4, on a en  $\text{op}(1, \text{op}(h, V))$  et  $\text{op}(2, \text{op}(h, V))$  de ce vecteur, respectivement l'indice  $n$  et la valeur d'un élément de ce vecteur.

On appelle  $g_i$  les 4 valeurs propres du système en excentricités:

```
> for l to npl do;
> g[l]:=vx[l][1]:
> od;
```

$g_1 := .00010802449768716235322 I$   
 $g_2 := .000017935389575491109393 I$   
 $g_3 := .000013091997026140007792 I$   
 $g_4 := .30701553267452553478 10^{-5} I$

On construit la matrice  $P$  de passage (ou des vecteurs propres) et la matrice  $e^{(D)t}$  (pour faciliter leur manipulation ultérieurement, on a préféré remplacer les expressions  $e^{(g_k t)}$  par des symboles  $eG_k$ ):

```
> PP:='PP': PP:=matrix(npl,npl): PPm1:=matrix(npl,npl):
> ExpD:=matrix(npl,npl):
> for k to npl do: print(op(3,op(1,vx[k][3]))):
> for l to npl do:
> ExpD[k,l]:=0: ll:=op(1,op(1,op(3,op(1,vx[k][3])))):
> PP[ll,k]:= op(2,op(1,op(3,op(1,vx[k][3])))):
```

```

> od: ExpD[k,k]:=eG[k]: od:
> print(PP):print(ExpD):
[4 = .0026674953944675523185, 1 = .30695755331641364419, 2 = -.95104758982733399253, 3 = .035755094936814944003]
[4 = .024387477836945310795 I, 1 = .63940413482000910381 I, 2 = .50351431815574310935 I, 3 = -.65625778169377757480 I]
[4 = -.13099940626113734020 I, 1 = .08527962873397042690 I, 2 = .07763094906832026625 I, 3 = 1.2117395561483676881 I]
[4 = .99068897602237082192 I, 1 = .006433718018462100846 I, 2 = .007248583957490107807 I, 3 = .15144779143659636351 I]
[
  .30695755331641364419    .63940413482000910381 I    .08527962873397042690 I    .006433718018462100846 I
  -.95104758982733399253    .50351431815574310935 I    .07763094906832026625 I    .007248583957490107807 I
  .035755094936814944003    -.65625778169377757480 I    1.2117395561483676881 I    .15144779143659636351 I
  .0026674953944675523185    .024387477836945310795 I    -.13099940626113734020 I    .99068897602237082192 I
]
[
  eG1    0    0    0
  0    eG2    0    0
  0    0    eG3    0
  0    0    0    eG4
]

```

Calcul de l'inverse de  $P$  :

```

> PPm1:=inverse(PP): print(PPm1);
[
  .66560403752577001693    -.83638496301095576376    .0068213491351340087558    .00075424244412470873539
  -1.1633360083676313694 I    -.37154198678780438124 I    .10505069172983838122 I    -.0057858382061138630946 I
  -.60422054597829591103 I    -.22307527004111691802 I    -.75536036523062174111 I    .12102892672919913939 I
  -.049466792438806094745 I    -.022603268571119108314 I    -.10244939361779130467 I    -.99325034630905857927 I
]

```

Vérification du produit  $P * P^{-1}$  qui devrait donner la matrice unité :



```
> print(multiply(PP,PPm1)):          # print(multiply(PPm1,PP)):
      [ .9999999999999999999          -.289 10-20          .468 10-20          -.4 10-21
        -.524 10-20          .9999999999999999999          .444 10-20          -.3 10-21
        .498 10-19          .136 10-19          1.0000000000000000000          -.1 10-19
        -.3 10-20          -.1 10-20          0          1.0000000000000000000 ]
```

```
> Digits:=10;
                               Digits := 10
```

Avec la valeur initiale des variables :

```
> initX:=vector([ 0.04698572124+I*0.01200385748,
> -0.00296003595+I*0.05542964254, -0.04595132376+I*0.00563791307,
> 0.00599977571+I*0.00669242413 ]);
```

*initX* := [.04698572124 + .01200385748 *I*, -.00296003595 + .05542964254 *I*, -.04595132376 + .00563791307 *I*, .00599977571 + .00669242413 *I*]

on obtient la solution :

```
> multiply(PPm1,initX): multiply(ExpD,%): SolX:=multiply(PP,%):
> SolX[1];
```

$$(.01026487256 - .01176482281 I) eG_1 + (.03735548309 + .02174320217 I) eG_2 + (-.0006572098286 + .001967114407 I) eG_3 + (.00002257540492 + .00005836370965 I) eG_4$$

Calcul de la solution pour les variables conjuguées :

```
> for k from 1 to npl do:
> SolXb[k]:=subs(I=-I,seq(eG[k1]=1/eG[k1],k1=1..npl),SolX[k]): od:
> SolXb[1];
```


$$\frac{.01026487256 + .01176482281 I}{eG_1} + \frac{.03735548309 - .02174320217 I}{eG_2} + \frac{-.0006572098286 - .001967114407 I}{eG_3} + \frac{.00002257540492 - .00005836370965 I}{eG_4}$$

Représentation graphique de la solution pour le "vecteur"  $e \exp(I\varpi)$  des planètes, sur 5 million d'années. Dans le plan complexe, on a une somme de vecteurs tournant avec les très longues périodes associées aux valeurs propres, d'où des épicycloïdes généralisées.

D'abord la solution x[1] pour Jupiter:

```
> Sol:=subs(seq(eG[k]=exp(g[k]*t),k=1..npl),SolX[1]);
```

```
Sol := (.01026487256 - .01176482281 I) e(.0001080244977 I t) + (.03735548309 + .02174320217 I) e(.00001793538957 I t)
+ (-.0006572098286 + .001967114407 I) e(.00001309199703 I t) + (.00002257540492 + .00005836370965 I) e(.307015532 10-5 I t)
> expand(convert(Sol,trig)):y(t):=coeff(select(has,%,I),I):x(t):=remove(has,%%,I):
> plot([x(t),y(t),t=0..5*10^6],numpoints=1000,scaling=CONSTRAINED);
```



Voici le développement limité de ce mouvement au voisinage de  $t=J2000$ , solution qui aurait été obtenue par une théorie à variations séculaires développée à l'ordre 4 des masses :


```
> Xvsop:=evalf(subs(t=T*1000,convert(series(x(t),t,5),polynom)),8);
> Yvsop:=evalf(subs(t=T*1000,convert(series(y(t),t,5),polynom)),8);
```

$$Xvsop := .046985721 + .00085498364 T - .000065843911 T^2 - .24500859 \cdot 10^{-5} T^3 + .58401604 \cdot 10^{-7} T^4$$

$$Yvsop := .012003857 + .0017703081 T + .000064977569 T^2 - .21922719 \cdot 10^{-5} T^3 - .66655684 \cdot 10^{-7} T^4$$


En comparant les courbes représentatives sur seulement 30 000 ans, on voit bien les limites de validité des théories à variations séculaires:

```
> plot([Xvsop,Yvsop,T=0..30],[x(t),y(t),t=0..30000],numpoints=1000,scaling=CONSTRAINED);
```




puis pour Saturne :

```
> Sol:=subs(seq(eG[k]=exp(g[k]*t),k=1..npl),SolX[2]):expand(convert(Sol  
> ,trig)): y(t):=coeff(select(has,%,I),I): x(t):=remove(has,%%,I):  
> plot([x(t),y(t),t=0..5*10^6],numpoints=1000,scaling=CONSTRAINED);
```




puis pour Uranus :

```
> Sol:=subs(seq(eG[k]=exp(g[k]*t),k=1..npl),SolX[3]):expand(convert(Sol  
> ,trig)): y(t):=coeff(select(has,%,I),I): x(t):=remove(has,%,I):  
> plot([x(t),y(t),t=0..5*10^6],numpoints=1000,scaling=CONSTRAINED);
```



et enfin pour Neptune :

```
> Sol:=subs(seq(eG[k]=exp(g[k]*t),k=1..npl),SolX[4]):expand(convert(Sol  
> ,trig)): y(t):=coeff(select(has,%,I),I): x(t):=remove(has,%,I):  
> plot([x(t),y(t),t=0..5*10^6],numpoints=1000,scaling=CONSTRAINED);
```



On procède de la même façon avec le système en inclinaisons :

```
> YY:=matrix(npl,npl): Digits:=20:  
> for k to npl do:
```

```

> for l to npl do:
> YY[k,l]:= evalf(subs(M=1,seq(m[k1]=TMAS[k1+dk],k1=1..npl),
> seq(n[k1]=AAM[k1+dk],k1=1..npl),seq(seq(alpha[k1,k2]=alf[k1+dk,k2+dk],
> k2=1..npl),k1=1..npl), LapLagy[k,l]),15):
> od: od:
> lambda[inc]:=[eigenvalues(YY)];

```

$\lambda_{inc} := [-.00012475653656261077367 I, -.000014079414575916449968 I, .16615550094228500000 10^{-19} I, -.32860884770123830040 10^{-5} I]$

Périodes en années, puis vecteurs propres (noter la valeur propre nulle):

```

> for k to npl do:
> print(evalf(2*Pi/lambda[inc][k]*I,10));
> od;

```

-50363.57596  
 -446267.5116  
 .3781509053 10<sup>21</sup>  
 -.1912056036 10<sup>7</sup>

```

> vy:=[eigenvectors(YY)];
> for l to npl do;
> f[l]:=vy[l][1];
> od;

```

$f_1 := .16615550094228500000 10^{-19} I$   
 $f_2 := -.00012475653656261077367 I$   
 $f_3 := -.32860884770123830040 10^{-5} I$   
 $f_4 := -.000014079414575916449968 I$

On a ainsi appelé  $\{f_k\}_{k=1..8}$  les valeurs propres de la matrice en inclinaisons.

valeurs initiales des variables d'inclinaison:

0.04061563384d0, 0.00682410142d0, 0.d0 , 0.01047042574d0, -0.00206561098d0, -0.00871747436d0, 0.00185915075d0, -0.01029147819d0,  
0.04563550461d0, 0.02882285775d0, 0.d0 , 0.01228449307d0, 0.01118377157d0, 0.01989147301d0, 0.00648617008d0, 0.01151683985d0

```
> initY:=vector([ -0.00206561098+I*0.01118377157,
> -0.00871747436+I*0.01989147301, 0.00185915075+I*0.00648617008,
> -0.01029147819+I*0.01151683985 ]);
```

```
initY := [-.00206561098 + .01118377157 I, -.00871747436 + .01989147301 I, .00185915075 + .00648617008 I, -.01029147819 + .01151683985 I]
```

matrice de passage

```
> QQ:=matrix(npl,npl): QQm1:=matrix(npl,npl):
> ExpDQ:=matrix(npl,npl):
> for k to npl do: print(op(3,op(1,vy[k][3]))):
> for l to npl do:
> ExpDQ[k,l]:=0: ll:=op(1,op(1,op(3,op(1,vy[k][3])))):
> QQ[ll,k]:= op(2,op(1,op(3,op(1,vy[k][3])))):
> od: ExpDQ[k,k]:=eF[k]: od:
> print(QQ):print(ExpDQ):
```

```
[4 = -.56921119765531153893 I, 1 = -.56921119765531565022 I, 2 = -.56921119765531551096 I, 3 = -.56921119765531295607 I]
```

```
[4 = .0045515201379227789091, 1 = .37190517353121971099, 2 = -.92735311357536503119, 3 = .041012538414777034854]
```

```
[4 = 1.1659483084508652803, 1 = -.11615815430905619198, 2 = -.11198054239776369193, 3 = .10840004410063602132]
```

```
[4 = -.11684932344958463578 I, 1 = -.054158655686664961603 I, 2 = -.044215320539539176171 I, 3 = .99256415625993840129 I]
```



$$\begin{bmatrix} -.56921119765531565022 I & .37190517353121971099 & -.11615815430905619198 & -.054158655686664961603 I \\ -.56921119765531551096 I & -.92735311357536503119 & -.11198054239776369193 & -.044215320539539176171 I \\ -.56921119765531295607 I & .041012538414777034854 & .10840004410063602132 & .99256415625993840129 I \\ -.56921119765531153893 I & .0045515201379227789091 & 1.1659483084508652803 & -.11684932344958463578 I \end{bmatrix}$$

$$\begin{bmatrix} eF_1 & 0 & 0 & 0 \\ 0 & eF_2 & 0 & 0 \\ 0 & 0 & eF_3 & 0 \\ 0 & 0 & 0 & eF_4 \end{bmatrix}$$

> QQm1:=inverse(QQ): print(QQm1);

$$\begin{bmatrix} 1.0818170815762258568 I & .43875308013039541709 I & .095180625458504336281 I & .14106637335311893459 I \\ .76327665716520968674 & -.77190035241054159545 & .0074056150100749042745 & .0012180802352568594325 \\ -.58999644097689865401 & -.23067928662105097355 & .048442201055985405173 & .77223352654196997289 \\ .58749922206262096721 I & .19452639640451870021 I & -.94731132134702428675 I & .16528570287988137700 I \end{bmatrix}$$

Solution pour les variables  $y$  et leurs conjuguées  $yb$

> multiply(QQm1,initY): multiply(ExpDQ,%): SolY:=multiply(QQ,%):  
> SolY[2];


$$\begin{aligned} & (-.0041747419747419794836 + .013130698655464300414 I) eF_1 + (-.0047792266616690014486 + .0062650684869632041578 I) eF_2 \\ & + (.00051821451150305472912 + .00022161419879442439775 I) eF_3 + (-.00028172023509207379696 + .00027409166877807103103 I) eF_4 \end{aligned}$$

> for k from 1 to npl do:  
> SolYb[k]:=subs(I=-I,seq(eF[k1]=1/eF[k1],k1=1..npl),SolY[k]): od:  
> SolYb[2];

$$\begin{aligned} & \frac{-.0041747419747419794836 - .013130698655464300414 I}{eF_1} + \frac{-.0047792266616690014486 - .0062650684869632041578 I}{eF_2} \\ & + \frac{.00051821451150305472912 - .00022161419879442439775 I}{eF_3} + \frac{-.00028172023509207379696 - .00027409166877807103103 I}{eF_4} \end{aligned}$$

Exemple de représentation de la solution, encore pour Jupiter :


```
> Digits:=10: Sol:=subs(seq(eF[k]=exp(f[k]*t),k=1..npl),SolY[1]):  
> expand(convert(Sol,trig)): y(t):=coeff(select(has,%,I),I):  
> x(t):=remove(has,%,I):  
> plot([x(t),y(t),t=0..5*10^6],numpoints=1000,scaling=CONSTRAINED);
```



Cette figure montre que le "pôle" moyen du plan orbital de Jupiter décrit une zone annulaire centrée sur un point fixe qui est le pôle du plan invariable du système solaire (le pôle d'une orbite désigne ici le point de la sphère céleste où le vecteur moment cinétique orbital perce cette sphère, )


Ce pôle du plan invariable correspond à la valeur propre nulle  
On a un résultat analogue pour le plan orbital moyen de Saturne:

```
> Sol:=subs(seq(eF[k]=exp(f[k]*t),k=1..npl),SolY[2]):expand(convert(Sol  
> ,trig)): y(t):=coeff(select(has,%,I),I): x(t):=remove(has,%%,I):  
> plot([x(t),y(t),t=0..5*10^6],numpoints=1000,scaling=CONSTRAINED);
```




puis celui d'Uranus

```
> Sol:=subs(seq(eF[k]=exp(f[k]*t),k=1..npl),SolY[3]):expand(convert(Sol  
> ,trig)): y(t):=coeff(select(has,%,I),I): x(t):=remove(has,%%,I):  
> plot([x(t),y(t),t=0..5*10^6],numpoints=1000,scaling=CONSTRAINED);
```



et de Neptune :

```
> Sol:=subs(seq(eF[k]=exp(f[k]*t),k=1..npl),SolY[4]):expand(convert(Sol
> ,trig)): y(t):=coeff(select(has,%,I),I): x(t):=remove(has,%%,I):
> plot([x(t),y(t),t=0..5*10^6],numpoints=1000,scaling=CONSTRAINED);
```



On pourrait ensuite calculer les termes séculaires de degré supérieur à 1 et se servir de cette solution pour déterminer par approximations successives les modifications des valeurs propres et du reste par l'introduction de ces termes non linéaires...

C(6.174)

Sauvegarde de la solution

```
> save SolX,SolXb,SolY,SolYb,g,f,
> cat(chemin,'SolXYLapLag_pla5_8.m');
```

## 2.4. Perturbations à très longues périodes dans les longitudes moyennes

```
> read cat(chemin,'SolXYLapLag_pla5_8.m'):
> read cat(chemin,'init_pla.txt'):
> dk:=4; Digits:=20:
```

$dk := 4$

On substitue les valeurs numériques des masses et des demi-grands axes dans les expressions  $Aq[k]$  représentant les second membres des équations de Lagrange pour la partie séculaire de la longitude moyenne, puis on y substitue la solution de Laplace-Lagrange (on opère cette substitution terme à terme pour accélérer le calcul) :

```
> for k1 from 1 to npl do:
> evalf(subs(M=1,seq(m[k1]=TMAS[k1+dk],k1=1..npl),
> seq(n[k1]=AAM[k1+dk],k1=1..npl),seq(seq(alpha[k1,k2]=alf[k1+dk,k2+dk],
> k2=1..npl),k1=1..npl), Aq[k1]),15): Sq[k1]:=expand(%): Tq:=0:
> for k2 from 1 to nops(Sq[k1]) do:
> Tq:=Tq+ subs(seq(x[k]=SolX[k],k=1..npl),
> seq(xb[k]=SolXb[k],k=1..npl), seq(y[k]=SolY[k],k=1..npl),
> seq(yb[k]=SolYb[k],k=1..npl),op(k2,Sq[k1])):
> od:
> Solq[k1]:=expand(Tq); print(k1):
> od:
```

1

2  
3  
4

L'intégration de ces termes par rapport au temps s'effectue en divisant chaque terme par la combinaison des valeurs propres présentes dans ce terme. Par exemple, un terme en  $eG_k^n eG_h^m eF_j^l$  s'intègre en le divisant par  $n g_k + m g_h + l f_l$  :

```
> for k from 1 to npl do:  Tq:=0:  print(k,nops(Solq[k])):
> for k1 from 1 to nops(Solq[k]) do:
> trm:=op(k1,Solq[k]):
> div:=0:
> for k2 from 1 to npl do:
> div:=div+degree(trm,eG[k2])*g[k2]+degree(trm,eF[k2])*f[k2]
> od:  # print(k1,trm,div):
> if abs(div)>10^(-8) then Tq:=Tq+evalf(trm/div,15) else Tq:=Tq+trm*t
> fi:
> od:
> Soq[k]:=Tq:
> od:
```

1, 50  
2, 50  
3, 50  
4, 50

Pour ne conserver que des termes utiles, on introduit ici une procédure TRONCn(u,d,v,f) qui tronque une série u dépendant de variables dont la liste est v , en éliminant les termes dont le coefficient (numérique) multiplié par un facteur f est inférieur en valeur absolue à  $10^{(-d)}$  . Pour obtenir le coefficient de chaque terme, on y substitue simplement 1 aux variables:

```

> TRONCn:=proc(u,d,v,f) local ux,k,sv;
> sv:=seq(op(k,v)=1,k=1..nops(v)):
> if type(u,'+') then
> ux:=0:
> for k from 1 to nops(u) do:
> if evalf(abs(f*subs(sv,op(k,u))),d+3)>=10^(-d) then :
> ux:=ux+op(k,u): fi:
> od:
> RETURN(sort(ux,v)):
> elif evalf(abs(f*subs(sv,u)),d+3)>=10^(-d) then : RETURN(u):
> else RETURN(0):
> fi;
> end:

```

exemple :

```
> Sq[1];
```

```

- .0000368149959354798
+ .365054660386260 10-5 y3 yb3 - .000477022814243369 y2 yb1 - .000128150189187149 x2 xb2 + .000512600756748592 y2 yb2
- .837378533118615 10-6 yb1 y4 + .949594668856527 10-6 y4 yb4 - .837378533118615 10-6 y1 yb4 + .690060831176423 10-7 xb1 x4
+ .690060831176423 10-7 x1 xb4 - .912636650965651 10-6 x3 xb3 - .000111256528292271 x1 xb1 - .000477022814243369 y1 yb2
+ .0000988661503744533 x2 xb1 + .000445026113169085 y1 yb1 + .0000988661503744533 x1 xb2 - .237398667214131 10-6 x4 xb4
+ .403221145755880 10-6 x1 xb3 - .325331281870973 10-5 yb1 y3 - .325331281870973 10-5 y1 yb3 + .403221145755880 10-6 xb1 x3

```



Troncature pour ne conserver que les coefficients plus grands que  $10^{-5}$  :

```
> TRONCn(Sq[1],5,varxy18,1);
- .0000368149959354798
- .000111256528292271 x1 xb1 + .0000988661503744533 x1 xb2 + .0000988661503744533 x2 xb1 - .000128150189187149 x2 xb2
- .000477022814243369 y1 yb2 - .000477022814243369 y2 yb1 + .000512600756748592 y2 yb2 + .000445026113169085 y1 yb1
```

Pour manipuler les séries semi-numériques dépendant des longues périodes, on introduit l'ensemble des variables:

```
> vareFG18:=[seq(eF[k],k=1..npl),seq(eG[k],k=1..npl),t];
vareFG18 := [eF1, eF2, eF3, eF4, eG1, eG2, eG3, eG4, t]
```

On évalue les constantes  $p_{0i}$  qui permettent d'éliminer le terme constant dans chaque équation  $dq_{0i}/dt$ ; ainsi la valeur moyenne des solutions  $q_{0i}$  sera nulle et la valeur moyenne de la dérivée  $dL_i/dt$  de chaque longitude moyenne sera le moyen mouvement moyen  $N_i$  introduit au début de la construction de la théorie générale.

```
> for k from 1 to npl do:
> Tconst:=TRONCn(coeff(select(has,Soq[k],t),t),6,vareFG18,1):
> print(k, Tconst, ' ', solve(N[k+dk]*p0[k] + Tconst=0, p0[k]));
> od:
1, -.000037300637353077320642, .000070419621649418967216
2, .00053155902176798090148, -.0024920828692519703600
3, .00020228116730784953073, -.0027049591234106313163
4, .00010286588031821342406, -.0026975528855155167607
```

On retrouve à peu près les valeurs des " poi " qui ont servi à initialiser les calculs des demi-grands axes en 2.2 . et qu'on rappelle ici (les différences viennent de la limitation actuelle à 4 planètes et à l'ordre 1 des masses:

constantes  $p_{0i}$

0.11795121D-5, 0.49815442D-5, -0.15429922D-5, 1.20304680D-5, 5.70597419D-5, -249.58418298D-5, -271.86256199D-5, -271.78052050D-5,


On peut alors représenter les variations à très longues périodes de la longitude moyenne des planètes :  
 Facteur de conversion (en secondes si Fa=206265, en ua si Fa=A[6] ...)

```
> Fa:=206265: ua:=149600000: Digits:=10:
```

Exemple de représentation pour les variations à très longue période de la longitude moyenne de Jupiter, exprimées en secondes sur 2 millions d'années:

```
> Soll:=TRONCn(remove(has,Soq[1],t),5,vareFG18,1):
> Sol:=combine(subs(seq(eG[k]=exp(g[k]*t),k=1..npl),seq(eF[k]=exp(f[k]*t),k=1..npl),Soll),exp):
> expand(convert(Sol,trig)): y(t):=remove(has,%,I)*Fa ;plot([t,y(t),t=-10^6..10^6]);
```


$$y(t) := 16.91082333 \sin(.0001247565366 t) - 6.055057309 \cos(.0001247565366 t) - 15.96194937 \sin(.00009008910812 t) + 82.86723330 \cos(.00009008910812 t) + 125.0831125 \cos(.484339255 \cdot 10^{-5} t) - 25.96644853 \sin(.484339255 \cdot 10^{-5} t)$$



On obtient ainsi des variations de très grande amplitude et à très longues périodes, ou des variations séculaires au voisinage de  $t = 0$  (J2000), avec  $T$  en milliers d'années, mais qui n'est valide que sur une durée de quelques milliers d'années seulement:

```
> Qvsop:=evalf(convert(series(y(t),t,5),polynom),8);  
> plot([t,Qvsop,t=-40000..30000],[t,y(t),t=-40000..30000]);
```


$$Qvsop := 201.89528 + .00007711119t - .29062329 \cdot 10^{-6} t^2 - .28228445 \cdot 10^{-11} t^3 + .16632292 \cdot 10^{-15} t^4$$



De même pour Saturne, on voit que l'amplitude des oscillations est 10 fois plus grande que pour Jupiter :

```
> Soll:=TRONCn(remove(has,Soq[2],t),5,vareFG18,1):
> Sol:=combine(subs(seq(eG[k]=exp(g[k]*t),k=1..npl),seq(eF[k]=exp(f[k]*
> t),k=1..npl),Soll),exp):expand(convert(Sol,trig)):
> y(t):=remove(has,%,I)*Fa ; plot([t,y(t),t=-10^6..10^6]);
```


$$\begin{aligned}
 y(t) := & 9.480001135 \sin(.0001106771220 t) - 6.155999337 \cos(.00001079332610 t) + 22.50410690 \sin(.00001486523425 t) \\
 & - 17.99774240 \cos(.00001486523425 t) - 4.256661293 \sin(.00001002184170 t) - 42.16748186 \sin(.0001247565366 t) \\
 & + 15.09840854 \cos(.0001247565366 t) + 56.48504341 \sin(.00009008910812 t) - 293.2448390 \cos(.00009008910812 t) \\
 & + 36.57273212 \sin(.00009493250067 t) + 15.24659402 \cos(.00009493250067 t) - 1986.635807 \cos(.484339255 10^{-5} t) \\
 & + 412.4127980 \sin(.484339255 10^{-5} t)
 \end{aligned}$$



```
> evalf(subs(t=T*1000,convert(series(y(t),t,5),polynom)),8);
      -2273.6894 + 6.6385246 T + 1.0294435 T2 - .00061342253 T3 - .00060093083 T4
```

puis pour Uranus (amplitude 30 fois plus importante que pour Jupiter):


```
> Soll:=TRONCn(remove(has,Soq[3],t),5,vareFG18,1):
> Sol:=combine(subs(seq(eG[k]=exp(g[k]*t),k=1..npl),seq(eF[k]=exp(f[k]*
> t),k=1..npl),Soll),exp):expand(convert(Sol,trig)):
> y(t):=remove(has,%,I)*Fa : plot([t,y(t),t=-10^6..10^6]);
```



```
> evalf(subs(t=T*1000,convert(series(y(t),t,5),polynom)),8);
7093.5945 - 26.680350 T - 1.4055482 T^2 + .031285735 T^3 + .00085996993 T^4
```

et enfin pour Neptune :

```
> Soll:=TRONCn(remove(has,Soq[4],t),5,vareFG18,1):
> Sol:=combine(subs(seq(eG[k]=exp(g[k]*t),k=1..npl),seq(eF[k]=exp(f[k]*
> t),k=1..npl),Soll),exp):expand(convert(Sol,trig)):
> y(t):=remove(has,%,I)*Fa : plot([t,y(t),t=-10^6..10^6]);
```



```
> evalf(subs(t=T*1000,convert(series(y(t),t,5),polynom)),8);
795.00823 - 1.3647721 T - .079605459 T2 + .18736519 10-5 T3 + .000053239236 T4
> Digits:=20;
```

### 3. Perturbations périodiques des éléments orbitaux des planètes à l'ordre 1 des masses

C<sub>26.2.2</sub>

Les perturbations périodiques d'ordre 1 sont classées par "inégalités"  $(p, p')$ , c'est-à-dire regroupant les termes qui sont en facteur d'une fonction périodique à courte période de la forme  $\exp(I(pL + p'L'))$  où  $L$  et  $L'$  sont les longitudes moyennes de 2 planètes. On montre ici comment calculer toutes les inégalités périodiques d'ordre 1 ayant la même caractéristique donnée  $ic$  (on a  $ic = p + p'$ ). Ces inégalités sont alors de la forme  $(p, ic - p)$  avec  $p$  arbitraire (pour  $ic = 3$ , on aura par exemple la grande inégalité en faisant  $p = -2$ ). La méthode proposée ici pourrait sans doute être la base du calcul des perturbations d'ordre 2 des masses, où interviennent 2 ou 3 planètes et pour lesquelles il faut combiner des termes de la fonction perturbatrice d'ordre 1 avec ceux de la solution d'ordre 1 ...

#### 3.1. Calcul d'une classe d'inégalités dans la fonction perturbatrice des planètes

On utilise la méthode de développement en coefficients de Laplace. On initialise le degré maximum souhaité pour les développements (degré en excentricités et inclinaisons); ici pour limiter les problèmes d'affichage, on choisit un degré maximum égal à 3 dans la fonction perturbatrice, permettant d'obtenir des perturbations au degré 2 (après les dérivées de cette fonction opérées dans le calcul des équations de Lagrange):

```
> Dmin:=0; Dmax:=3;
Dmin := 0
Dmax := 3
```

```

> deb0:=time():
> inineg:=IniINEGLap(Dmin,Dmax):
> XnmT:= op(1,inineg):
> COSSYm1T:=TRONC(COSSYm1,Dmax,[Y[i],Yb[i],Y[j],Yb[j]]):
> for k from 0 to Dmax/2 do: DELTACOSSYmT.k:=op(k+2,inineg) od:
> s:='s': k:='k': m:='m':
> Usn:=expand(sum(sum(Ca(s,k,m,alpha)*beta^(k)*(beta^2-1)^m*W^k,k=0..Dma
> x/2),m=0..Dmax)):
> DbetaWn:=combine(Usn,power): nops(%);
> print(time()-deb0):

```

20  
.010

Puis on fixe la caractéristique de l'inégalité (on pourrait aussi faire une boucle pour déterminer toutes les inégalités de caractéristique comprise entre -Dmax et +Dmax)

```

> ic:=-2; p:='p'; pp:=p; qq:=-p+ic;

```

$$ic := -2$$

$$p := p$$

$$pp := p$$

$$qq := -p - 2$$

Calcul de la partie indirecte des perturbations concernant cette inégalité :

```

> ingINDi:=INEG(pp,qq,Dmin,Dmax,1,-2,COSSYm1T,XnmT):
> PRINTRONC(ingINDi,2,varxyij):

```



$$\begin{aligned} & \left(\frac{1}{16} \begin{pmatrix} 1 & p+2=1 \\ 0 & otherwise \end{pmatrix} + \frac{3}{16} \begin{pmatrix} 1 & p+2=-1 \\ 0 & otherwise \end{pmatrix}\right) x_i^2 - \frac{3}{2} \begin{pmatrix} 1 & p+1=1 \\ 0 & otherwise \end{pmatrix} x_i x_j \\ & + \left(\frac{27}{16} \begin{pmatrix} 1 & p=1 \\ 0 & otherwise \end{pmatrix} + \frac{1}{16} \begin{pmatrix} 1 & p=-1 \\ 0 & otherwise \end{pmatrix}\right) x_j^2 + \frac{1}{2} \begin{pmatrix} 1 & p+2=1 \\ 0 & otherwise \end{pmatrix} y_i^2 \\ & - C_i C_j \begin{pmatrix} 1 & p+1=0 \\ 0 & otherwise \end{pmatrix} y_i y_j + \frac{1}{2} \begin{pmatrix} 1 & p=-1 \\ 0 & otherwise \end{pmatrix} y_j^2 \end{aligned}$$

- > `ingINDj:=INEG(pp,qq,Dmin,Dmax,-2,1,COSSYm1T,XnmT):`
- > `PRINTRONC(ingINDj,2,varxyij):`

$$\begin{aligned} & \left(\frac{1}{16} \begin{pmatrix} 1 & p+2=1 \\ 0 & otherwise \end{pmatrix} + \frac{27}{16} \begin{pmatrix} 1 & p+2=-1 \\ 0 & otherwise \end{pmatrix}\right) x_i^2 - \frac{3}{2} \begin{pmatrix} 1 & p+1=-1 \\ 0 & otherwise \end{pmatrix} x_i x_j \\ & + \left(\frac{3}{16} \begin{pmatrix} 1 & p=1 \\ 0 & otherwise \end{pmatrix} + \frac{1}{16} \begin{pmatrix} 1 & p=-1 \\ 0 & otherwise \end{pmatrix}\right) x_j^2 + \frac{1}{2} \begin{pmatrix} 1 & p+2=1 \\ 0 & otherwise \end{pmatrix} y_i^2 \\ & - C_i C_j \begin{pmatrix} 1 & p+1=0 \\ 0 & otherwise \end{pmatrix} y_i y_j + \frac{1}{2} \begin{pmatrix} 1 & p=-1 \\ 0 & otherwise \end{pmatrix} y_j^2 \end{aligned}$$

- > `UiPerturBParj:=alpha*ingINDi:`
- > `UjPerturBParj:=1/alpha^2*ingINDj:`

Calcul de la partie directe en coefficients de Laplace (affichage du premier terme seulement):

- > `deb:=time():`
- > `UD1:=INVDLAP(pp,qq,Dmin,Dmax,DbetaWn,DELTACOSSYmT,XnmT):`
- > `UDa:=sort(collect(UD1,varxyij,distributed),varxyij): op(1,%):`
- > `print(nops(%),"duree=",time()-deb):`

$$\begin{aligned}
 & \left( -\frac{5}{8}p - \frac{5}{4} + \frac{1}{2}|p+2|(p+2) + \frac{1}{2}(p+2)^2 - \frac{3}{8}|p+2| + \frac{1}{8}|p+2|^2 \right) \\
 & \quad \times (\text{Ca}(p+2, 0, 0, \alpha) + \text{Ca}(p+2, 0, 2, \alpha) - \text{Ca}(p+2, 0, 1, \alpha) - \text{Ca}(p+2, 0, 3, \alpha)) \\
 & + \left( -\frac{5}{8}p - 2 + \frac{1}{2}(2+|p+2|)(p+2) + \frac{1}{2}(p+2)^2 - \frac{3}{8}|p+2| + \frac{1}{8}(2+|p+2|)^2 \right) \\
 & \quad \times (-2\text{Ca}(p+2, 0, 2, \alpha) + 3\text{Ca}(p+2, 0, 3, \alpha) + \text{Ca}(p+2, 0, 1, \alpha)) \\
 & + \left( -\frac{5}{8}p - \frac{11}{4} + \frac{1}{2}(4+|p+2|)(p+2) + \frac{1}{2}(p+2)^2 - \frac{3}{8}|p+2| + \frac{1}{8}(4+|p+2|)^2 \right) (\text{Ca}(p+2, 0, 2, \alpha) - 3\text{Ca}(p+2, 0, 3, \alpha)) \\
 & + \left( -\frac{5}{8}p - \frac{7}{2} + \frac{1}{2}(6+|p+2|)(p+2) + \frac{1}{2}(p+2)^2 - \frac{3}{8}|p+2| + \frac{1}{8}(6+|p+2|)^2 \right) \text{Ca}(p+2, 0, 3, \alpha) x_i^2 \\
 & \quad 6, \text{ "duree"} = 0
 \end{aligned}$$

Rappelons que les fonctions  $\text{Ca}(-, -, -, \alpha)$  sont des fonctions plus ou moins compliquées des coefficients de Laplace qu'on pourra évaluer quand on en aura besoin en faisant `eval(subs(Ca=Calf, expx))` ou `eval(subs(Ca=Calfa, expx))`

D'où les fonctions perturbatrices des cas "intérieur" et "extérieur" (affichage du quatrième terme pour exemple):

- > `UDai:=eval(sort(collect(UDa-UiPerturBParj,varxyij,distributed),varxyij)):op(4,%);`
- > `eval(subs(Ca=Calf,sort(collect(UDa-UiPerturBParj,varxyij,distributed),varxyij))):op(4,%);`
- > `eval(subs(Ca=Calfa,sort(collect(UDa-UiPerturBParj,varxyij,distributed),varxyij))):op(4,%);`

$$\begin{aligned}
 & (\text{Ca}(p+1, 1, 0, \alpha) - \frac{1}{2}\alpha \left( \begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases} \right)) y_i^2 \\
 & (\text{Cl}(|p+1|, 1, 0) \alpha^{(1+|p+1|)} \text{F}\left(\left[\frac{3}{2}, \frac{3}{2} + |p+1|\right], [1 + |p+1|], \alpha^2\right) - \frac{1}{2}\alpha \left( \begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases} \right)) y_i^2 \\
 & \left( \frac{1}{2} \frac{\text{pochhammer}\left(\frac{3}{2}, |p+1|\right) \text{pochhammer}\left(\frac{3}{2} + |p+1|, 0\right) \alpha^{(1+|p+1|)} \text{F}\left(\left[\frac{3}{2}, \frac{3}{2} + |p+1|\right], [1 + |p+1|], \alpha^2\right)}{|p+1|! \text{pochhammer}(1 + |p+1|, 0)} - \frac{1}{2}\alpha \left( \begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases} \right) \right) y_i^2
 \end{aligned}$$

```
> UDaj:=eval(sort(collect(UDa-UjPerturBPar, varxyij, distributed), varxyij)):op(4,%);
```

$$\left( Ca(p+1, 1, 0, \alpha) - \frac{1}{2} \begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases} \frac{1}{\alpha^2} \right) y_i^2$$

On a enfin les fonctions perturbatrices des deux planètes pour l'inégalité choisie :

```
> UPai:=K*m[j]/a[j]*UDai*eL[i]^pp*eL[j]^qq:
```

```
> UPaj:=K*m[i]/a[j]*UDaj*eL[i]^pp*eL[j]^qq:
```

```
> save UPai, UPaj, cat(chemin,'UPaijD'.Dmax.'('.ic.').m');
```

Calculer une [autre inégalité](#) au même degré, ou à un [autre degré](#)

### 3.2. Calcul des équations de Lagrange concernant une classe d'inégalités

On commence par lire le fichier contenant la classe d'inégalités choisie :

```
> Dmax:=3: ic:=-2: p:='p': pp:=p: qq:=ic-p:
```

```
> read cat(chemin,'UPaijD'.Dmax.'('.ic.').m');
```

Equations de Lagrange pour une planète **intérieure** (indice  $i$ ) perturbée par une planète **extérieure** (indice  $j$ )

```
> DVai:=EqLag(expand(UPai), "interieur", i, j):
```

```
> seq(nops(DVai[k]), k=1..4);
```

6, 6, 1, 1

Exemple de terme où l'on voit des dérivées des fonctions  $Ca(-, -, -, \alpha)$  par rapport à  $\alpha$  non encore explicitées dans l'équation de Lagrange  $dq/dt$  (terme  $D_4(Ca)(-, -, -, \alpha)$ ):

```
> op(4,DVai[2]);
```

$$\left( \frac{K m_j eL_i^p \left( \begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases} \right)}{n_i a_i a_j^2 eL_j^p eL_j^2} - 2 \frac{K m_j eL_i^p D_4(Ca)(p+1, 1, 0, \alpha)}{n_i a_i a_j^2 eL_j^p eL_j^2} - \frac{1}{2} \frac{K m_j eL_i^p \alpha \left( \begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases} \right)}{n_i a_i^2 \phi_i a_j eL_j^p eL_j^2} + \frac{K m_j eL_i^p Ca(p+1, 1, 0, \alpha)}{n_i a_i^2 \phi_i a_j eL_j^p eL_j^2} \right) y_i^2$$

Equations de Lagrange pour une planète **extérieure** (indice  $j$ ) perturbée par une planète **intérieure** (indice  $i$ )

```
> DVaj:=EqLag(expand(UPaj), "exterieur", i, j);
> seq(nops(DVaj[k]), k=1..4);
```

6, 6, 1, 1

Pour évaluer les dérivées par rapport à  $\alpha$ , on introduit la procédure :

```
> DCalf:=proc(s,k,m,alpha) local h,j: h:=(2*k+1)/2;
> j:=abs(s):Cl(j,k,m)*diff(alpha^(k+2*m+j)*hypergeom([h+m,h+j+m],[1+j+m]
> ,alpha^2),alpha);
> end;
```

$DCalf := \text{proc}(s, k, m, \alpha) \text{ local } h, j; h := k + 1/2; j := \text{abs}(s); Cl(j, k, m) \times \text{diff}(\alpha^{(k+2 \times m+j)} \times F([h+m, h+j+m], [1+j+m], \alpha^2), \alpha) \text{ end}$

Et pour le terme :

```
> collect(sort(collect(op(4,DVai[2]), varxyij, distributed, factor), varxyi
> j), Ca);
```

$$\frac{K m_j eL_i^p y_i^2 Ca(p+1, 1, 0, \alpha)}{a_j n_i a_i^2 eL_j^p eL_j^2 \phi_i} - 2 \frac{K m_j eL_i^p y_i^2 D_4(Ca)(p+1, 1, 0, \alpha)}{a_i n_i a_j^2 eL_j^p eL_j^2} - \frac{1}{2} \frac{K m_j eL_i^p (-2 \left( \begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases} \right) a_i \phi_i + \alpha \left( \begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases} \right) a_j) y_i^2}{n_i a_i^2 a_j^2 eL_j^p eL_j^2 \phi_i}$$

on a alors :

```
> eval(subs(D[4](Ca)=DCalf, Ca=Calf, %));
```

$$\frac{K m_j e L_i^p y_i^2 \text{Cl}(|p+1|, 1, 0) \alpha^{(1+|p+1|)} F\left(\frac{3}{2}, \frac{3}{2} + |p+1|, [1 + |p+1|], \alpha^2\right)}{a_j n_i a_i^2 e L_j^p e L_j^2 \phi_i} - 2 \frac{K m_j e L_i^p y_i^2 \text{Cl}(|p+1|, 1, 0)}{a_i n_i a_j^2 e L_j^p e L_j^2}$$

$$\times \left( \frac{\alpha^{(1+|p+1|)} (1 + |p+1|) F\left(\frac{3}{2}, \frac{3}{2} + |p+1|, [1 + |p+1|], \alpha^2\right)}{\alpha} + 2 \frac{\alpha^{(1+|p+1|)} \left(\frac{9}{4} + \frac{3}{2} |p+1|\right) F\left(\frac{5}{2}, \frac{5}{2} + |p+1|, [2 + |p+1|], \alpha^2\right) \alpha}{1 + |p+1|} \right)$$

$$- \frac{1}{2} \frac{K m_j e L_i^p \left(-2 \left(\begin{matrix} 1 & p+2=1 \\ 0 & \text{otherwise} \end{matrix}\right) a_i \phi_i + \alpha \left(\begin{matrix} 1 & p+2=1 \\ 0 & \text{otherwise} \end{matrix}\right) a_j\right) y_i^2}{n_i a_i^2 a_j^2 e L_j^p e L_j^2 \phi_i}$$

On a alors finalement les équations relatives à cette inégalité complètement évaluées pour les planètes  $i$  et  $j$  :

```
> for kv from 1 to 4 do:
> collect(sort(collect(DVai[kv], varxyij, distributed, factor), varxyij), Ca):
> eval(subs(D[4](Ca)=DCalf, Ca=Calf, %)):
> subs(alpha=alpha[i, j], K=n[i]^2*a[i]^3/(M+m[i]), a[i]=alpha[i, j]*a[j], %):
> if kv=1 then: Da[i]:=:%: elif kv=2 then: Dq[i]:=:%: elif kv=3
> then: Dx[i]:=:%: else: Dy[i]:=:%: fi:
> collect(sort(collect(DVaj[kv], varxyij, distributed, factor), varxyij), Ca):
> eval(subs(D[4](Ca)=DCalf, Ca=Calf, %)):
> subs(alpha=alpha[i, j], K=n[j]^2*a[j]^3/(M+m[j]), a[i]=alpha[i, j]*a[j], %):
> if kv=1 then: Da[j]:=:%: elif kv=2 then: Dq[j]:=:%: elif kv=3
> then: Dx[j]:=:%: else: Dy[j]:=:%: fi:
> od:
> nops(Da[i]); nops(Dq[i]); nops(Dx[i]); nops(Dy[i]); nops(Da[j]);
> nops(Dq[j]); nops(Dx[j]); nops(Dy[j]);
```

16

23

1  
1  
16  
26  
1  
1

On y substitue encore les fonctions  $\phi_i$ ,  $\phi_j$ ,  $C_i$  et  $C_j$  par leur développement :

```
> Dm:=Dmax:
> subs(phi[i]=sqrt(1-x[i]*xb[i]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Da[i]):
> Da[i]:=mtaylor(%,varxyij,Dm):
> subs(phi[j]=sqrt(1-x[j]*xb[j]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Da[j]):
> Da[j]:=mtaylor(%,varxyij,Dm):
> subs(phi[i]=sqrt(1-x[i]*xb[i]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dq[i]):
> Dq[i]:=mtaylor(%,varxyij,Dm):
> subs(phi[j]=sqrt(1-x[j]*xb[j]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dq[j]):
> Dq[j]:=mtaylor(%,varxyij,Dm):
> subs(phi[i]=sqrt(1-x[i]*xb[i]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dx[i]):
> Dx[i]:=mtaylor(%,varxyij,Dm):
> subs(phi[j]=sqrt(1-x[j]*xb[j]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dx[j]):
> Dx[j]:=mtaylor(%,varxyij,Dm):
```

```
> subs(phi[i]=sqrt(1-x[i]*xb[i]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dy[i]):
> Dy[i]:=mtaylor(%,varxyij,Dm):
> subs(phi[j]=sqrt(1-x[j]*xb[j]),C[i]=sqrt(1-y[i]*yb[i]),
> C[j]=sqrt(1-y[j]*yb[j]),Dy[j]):
> Dy[j]:=mtaylor(%,varxyij,Dm):
```

Pour intégrer cette inégalité périodique, on procède comme pour l'intégration effectuée dans la partie à longues périodes de la longitude moyenne, en divisant chaque terme par sa fréquence, c'est-à-dire la combinaison des moyens mouvements moyens correspondant à l'inégalité; on procède aussi à la double intégration du terme  $n_i = N_i \Delta p_i$  qu'il faut ajouter à chaque équation  $dq_i/dt$ , où  $\Delta p_i = -(3/2) \Delta a_i/A_i$ :

```
> div:=I*(pp*N[i]+qq*N[j]);


$div := I(p N_i + (-p - 2) N_j)$


> Dai:=Da[i]/div: Dqi:=(N[i]*(-3/2)*Dai/A[i]+Dq[i])/div :
> Dxi:=Dx[i]/div: Dyi:=Dy[i]/div:
> Daj:=Da[j]/div: Dqj:=(N[j]*(-3/2)*Daj/A[j]+Dq[j])/div :
> Dxj:=Dx[j]/div: Dyj:=Dy[j]/div:
```

Exemple de terme dans  $\Delta a_i$  :

```
> op(5,collect( expand(Dai),varxyij,distributed));
```

$$\left( -\frac{n_i \alpha_{i,j}^3 a_j m_j e L_i^p p \begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases}}{(p N_i - N_j p - 2 N_j) (M + m_i) e L_j^p e L_j^2} + 2 \frac{n_i \alpha_{i,j}^3 a_j m_j e L_i^p p \text{Cl}(|p+1|, 1, 0) \alpha_{i,j}^{|p+1|} F\left(\left[\frac{3}{2}, \frac{3}{2} + |p+1|\right], [1 + |p+1|], \alpha_{i,j}^2\right)}{(p N_i - N_j p - 2 N_j) (M + m_i) e L_j^p e L_j^2} \right) y_i^2$$

puis dans  $\Delta q_i$  :

```
> op(5,collect( expand(Dqi),varxyij,distributed));
```

$$\begin{aligned}
 & \left( 2 \frac{I n_i \alpha_{i,j}^2 m_j eL_i^p \text{Cl}(|p+1|, 1, 0) \alpha_{i,j}^{|p+1|} F\left(\frac{3}{2}, \frac{3}{2} + |p+1|, [1 + |p+1|], \alpha_{i,j}^2\right) |p+1|}{(p N_i - N_j p - 2 N_j) (M + m_i) eL_j^p eL_j^2} \right. \\
 & + 6 \frac{I n_i \alpha_{i,j}^4 m_j eL_i^p \text{Cl}(|p+1|, 1, 0) \alpha_{i,j}^{|p+1|} F\left(\frac{5}{2}, \frac{5}{2} + |p+1|, [2 + |p+1|], \alpha_{i,j}^2\right) |p+1|}{(p N_i - N_j p - 2 N_j) (M + m_i) eL_j^p eL_j^2 (1 + |p+1|)} \\
 & + \frac{I n_i \alpha_{i,j}^2 m_j eL_i^p \text{Cl}(|p+1|, 1, 0) \alpha_{i,j}^{|p+1|} F\left(\frac{3}{2}, \frac{3}{2} + |p+1|, [1 + |p+1|], \alpha_{i,j}^2\right)}{(p N_i - N_j p - 2 N_j) (M + m_i) eL_j^p eL_j^2} \\
 & + 3 \frac{I N_i n_i \alpha_{i,j}^3 a_j m_j eL_i^p p \text{Cl}(|p+1|, 1, 0) \alpha_{i,j}^{|p+1|} F\left(\frac{3}{2}, \frac{3}{2} + |p+1|, [1 + |p+1|], \alpha_{i,j}^2\right)}{(p N_i - N_j p - 2 N_j)^2 A_i (M + m_i) eL_j^p eL_j^2} \\
 & + 9 \frac{I n_i \alpha_{i,j}^4 m_j eL_i^p \text{Cl}(|p+1|, 1, 0) \alpha_{i,j}^{|p+1|} F\left(\frac{5}{2}, \frac{5}{2} + |p+1|, [2 + |p+1|], \alpha_{i,j}^2\right)}{(p N_i - N_j p - 2 N_j) (M + m_i) eL_j^p eL_j^2 (1 + |p+1|)} \\
 & \left. - \frac{1}{2} \frac{I n_i \alpha_{i,j}^2 m_j eL_i^p \left(\begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases}\right)}{(p N_i - N_j p - 2 N_j) (M + m_i) eL_j^p eL_j^2} - \frac{3}{2} \frac{I N_i n_i \alpha_{i,j}^3 a_j m_j eL_i^p p \left(\begin{cases} 1 & p+2=1 \\ 0 & \text{otherwise} \end{cases}\right)}{(p N_i - N_j p - 2 N_j)^2 A_i (M + m_i) eL_j^p eL_j^2} \right) y_i^2
 \end{aligned}$$

```

> save Dai,Dqi,Dxi,Dyi, Daj,Dqj,Dxj,Dyj,
> cat(chemin,'DeltaV'.Dmax.'('.ic.').m'):

```

autre inégalité

```

> Dmax,ic;

```

3, -2



### 3.3. Particularisations à un couple de planètes

```
> Dmax:=3:      ic:=-2:  p:='p':  pp:=p:  qq:=ic-p:
> read cat(chemin,'DeltaV'.Dmax.'(\.ic.').m'):
```

Particularisation pour le couple Jupiter Saturne et l'inégalité  $2L_5 - 4L_6$  (correspondant à  $ic = -2$  et  $p = 2$ )

```
> ki:=5; kj:=6; vp:=2;

                               ki := 5
                               kj := 6
                               vp := 2
> Da[ki]:=eval(subs(Cl=Clap,i=ki,j=kj,p=vp,alpha[ki,kj]=alf[ki,kj],Dai)
> ):
> Dq[ki]:=eval(subs(Cl=Clap,i=ki,j=kj,p=vp,alpha[ki,kj]=alf[ki,kj],Dqi)
> ):
> Dx[ki]:=eval(subs(Cl=Clap,i=ki,j=kj,p=vp,alpha[ki,kj]=alf[ki,kj],Dxi)
> ):
> Dy[ki]:=eval(subs(Cl=Clap,i=ki,j=kj,p=vp,alpha[ki,kj]=alf[ki,kj],Dyi)
> ):
> Da[kj]:=eval(subs(Cl=Clap,i=ki,j=kj,p=vp,alpha[ki,kj]=alf[ki,kj],Daj)
> ):
> Dq[kj]:=eval(subs(Cl=Clap,i=ki,j=kj,p=vp,alpha[ki,kj]=alf[ki,kj],Dqj)
> ):
> Dx[kj]:=eval(subs(Cl=Clap,i=ki,j=kj,p=vp,alpha[ki,kj]=alf[ki,kj],Dxj)
> ):
> Dy[kj]:=eval(subs(Cl=Clap,i=ki,j=kj,p=vp,alpha[ki,kj]=alf[ki,kj],Dyj)
> ):
> factor(Da[5]);
```

$$\frac{eL_5^2 m_6 a_6 n_5}{(M + m_5) eL_6^4} (2.442038185 x_5^2 - 8.261704804 x_6 x_5 + 6.902143932 x_6^2 + 1.007728137 y_5^2 - 2.015456275 y_6 y_5 + 1.007728137 y_6^2)$$

> factor(Dq[5]);

$$\frac{I n_5 m_6 eL_5^2}{(M + m_5) eL_6^4} (1.808785295 a_6 x_5^2 - 6.119335174 a_6 x_6 x_5 + 5.112326464 a_6 x_6^2 + .7464108656 a_6 y_5^2 - 1.492821731 a_6 y_6 y_5$$

$$+ .7464108656 a_6 y_6^2 + 9.153100659 x_5^2 - 25.33988017 x_6 x_5 + 16.48029195 x_6^2 + 4.584734354 y_5^2 - 9.632150444 y_6 y_5 + 5.047416091 y_6^2)$$

> factor(Da[6]);

$$-\frac{eL_5^2 m_5 a_6 n_6}{(M + m_6) eL_6^4} (16.47329117 x_5^2 - 55.73109773 x_6 x_5 + 46.55988893 x_6^2 + 6.797845803 y_5^2 - 13.59569161 y_6 y_5 + 6.797845803 y_6^2)$$

> factor(Dq[6]);

$$\frac{-I eL_5^2 m_5 n_6}{(M + m_6) eL_6^4} (2.675361215 a_6 x_5^2 - 9.051064282 a_6 x_6 x_5 + 7.561605010 a_6 x_6^2 + 1.104010900 a_6 y_5^2 - 2.208021800 a_6 y_6 y_5$$

$$+ 1.104010900 a_6 y_6^2 + 22.98746443 x_5^2 - 67.43671220 x_6 x_5 + 47.72657424 x_6^2 + 10.96921253 y_5^2 - 22.78815578 y_6 y_5 + 11.81894326 y_6^2)$$

Les perturbations concernant cette inégalité dans les solutions  $\Delta x_5$ ,  $\Delta y_5$ ,  $\Delta x_6$  et  $\Delta y_6$  sont nulles au degré 2 en excentricités et inclinaisons auquel on a limité ici les développements:

- > Dx[5]; 0
- > Dy[5]; 0
- > Dx[6]; 0
- > Dy[6]; 0

Procédure remplaçant toutes les constantes physiques par leur valeur :

```
> npl,dk;
```

4, 4

```
> Numeriq:=proc(ser) local k1,k2:
> collect(expand(evalf(subs(M=1,Ca=CalF,seq(m[k1]=TMAS[k1],k1=1+dk..npl+dk),
> seq(a[k1]=A[k1],k1=1+dk..npl+dk),
> seq(n[k1]=AAM[k1],k1=1+dk..npl+dk),
> seq(seq(alpha[k1,k2]=alf[k1,k2],k2=1+dk..npl+dk),k1=1+dk..npl+dk),ser),15)),
> [seq(eL[k1],k1=1+dk..npl+dk)],distributed):
> end:
> Numeriq(Da[5]);
```

$$\left( (.0035295834577198095329 x_5^2 - .011940999443370511342 x_6 x_5 + .0099759672838815361828 x_6^2 + .0014565130821147965943 y_5^2 - .0029130261642295883386 y_6 y_5 + .0014565130821147965943 y_6^2) eL_5^2 \right) / eL_6^4$$

```
> Numeriq(Dq[5]);
```

$$\left( (.0039988994512673344897 I x_5^2 - .012677687696024999997 I x_6 x_5 + .0098820300829214120312 I x_6^2 + .0017723498126566946651 I y_5^2 - .0036146892016068442409 I y_6 y_5 + .0018423393889501531681 I y_6^2) eL_5^2 \right) / eL_6^4$$

```
> Numeriq(Da[6]);
```

$$\left( (-.031965836331367236060 x_5^2 + .10814421543280830812 x_6 x_5 - .090347810517460361472 x_6^2 - .013190978299590876112 y_5^2 + .026381956599181752225 y_6 y_5 - .013190978299590876112 y_6^2) eL_5^2 \right) / eL_6^4$$

```
> Numeriq(Dq[6]);
```

$$\left( (-.0098599370119545981100 I x_5^2 + .031258901020985278038 I x_6 x_5 - .024365752466460819546 I x_6^2 - .0043700167330804745931 I y_5^2 + .0089126041502095716418 I y_6 y_5 - .0045425874171290970488 I y_6^2) eL_5^2 \right) / eL_6^4$$

Substitution de la solution à longues périodes dans cette inégalité :

```
> npl;dk;

                                4
                                4

> SubsLP:=proc(ser) local k,k1,s,Tq:  s:=expand(ser):
> if type(s,`+`) then:
> Tq:=0:
> for k1 from 1 to nops(s) do:
> Tq:=Tq+ subs(seq(x[k+dk]=SolX[k],k=1..npl),
> seq(xb[k+dk]=SolXb[k],k=1..npl),
> seq(y[k+dk]=SolY[k],k=1..npl),
> seq(yb[k+dk]=SolYb[k],k=1..npl),op(k1,s)):
> od:
> else Tq:=subs(seq(x[k+dk]=SolX[k],k=1..npl),
> seq(xb[k+dk]=SolXb[k],k=1..npl),
> seq(y[k+dk]=SolY[k],k=1..npl),
> seq(yb[k+dk]=SolYb[k],k=1..npl),ser):
> fi:
> RETURN(expand(Tq)):
> end:

> vareGFL18:=[seq(eF[k],k=1..npl),seq(eG[k],k=1..npl),seq(eL[k],k=1+dk.
> .npl+dk),t];
                                vareGFL18 := [eF1, eF2, eF3, eF4, eG1, eG2, eG3, eG4, eL5, eL6, eL7, eL8, t]
```

Exemple de substitution dans  $\Delta_{q_6}$ , coefficients en secondes de degré:

```
> SubsLP(Numeriq(Dq[6]))*206265:
```

affichage des termes plus grands que 0"1 dans  $\Delta q_6$ :

> TRONCn(%, 1, vareGFL18, 1);

$$\begin{aligned}
 & -1.0894178807622148431 \frac{eF_2^2 eL_5^2}{eL_6^4} + 2.3214850205109025807 \frac{I eG_1^2 eL_5^2}{eL_6^4} - 16.968735844738891415 \frac{eG_1^2 eL_5^2}{eL_6^4} \\
 & + 1.2020039220689101189 \frac{eG_1 eG_2 eL_5^2}{eL_6^4} + .16683301184909718149 \frac{I eG_1 eG_3 eL_5^2}{eL_6^4} - .28412440022957591104 \frac{eG_1 eG_3 eL_5^2}{eL_6^4} \\
 & + 3.5525823570055924031 \frac{I eG_1 eG_2 eL_5^2}{eL_6^4} + .11859515569158642516 \frac{eG_2^2 eL_5^2}{eL_6^4}
 \end{aligned}$$

# Contents

<b>1</b>	<b>Initialisations</b>	<b>3</b>
<b>2</b>	<b>Perturbations à très longues périodes des orbites des planètes</b>	<b>5</b>
2.1	Système formel de Laplace Lagrange à l'ordre 1 des masses . . . . .	5
2.2	Application aux planètes du système solaire . . . . .	15
2.3	Solution numérique du système de Laplace-Lagrange . . . . .	20
2.4	Perturbations à très longues périodes dans les longitudes moyennes . . . . .	38
<b>3</b>	<b>Perturbations périodiques des éléments orbitaux des planètes à l'ordre 1 des masses</b>	<b>47</b>
3.1	Calcul d'une classe d'inégalités dans la fonction perturbatrice des planètes . . . . .	47
3.2	Calcul des équations de Lagrange concernant une classe d'inégalités . . . . .	51
3.3	Particularisations à un couple de planètes . . . . .	57